

カシオパソコン

PB-400 入門書

やさしく覚えて、すぐに役立つ取扱説明書つき



カシオ計算機株式会社







——はじめに——

本書はこれからBASICプログラムを始めようとする方にも、既にBASICを知っていてPB-400をフルに活用しようとなさる方にも、わかりやすく、すぐに使えるように説明してあります。

BASICプログラムを初めて習う方は第1章から順にお読みになり、プログラムの基本をしっかりと覚えてください。特に第3章では、プログラムの作り方やコマンドの説明をよくお読みください。なお、第3章ではプログラムの流れにそって説明しておりますので、各コマンドの書式や詳しい説明については第4章のコマンドリファレンス編を参照してください。

既にBASICを知っている方は、第1章、第2章で基本的な操作を覚えただ後は、第4章のコマンドリファレンスを読みながらPB-400をお使いください。

すぐにプログラムを入力して使いたい方は、第5章の「実用ライブラリー」や、第3章の5によりPB-100のプログラムを使うことができます。

では、この本を参考にしてPB-400を有効に使ってください。

ご使用の前に

この計算機は、カシオの高度な電子技術と品質管理のもとで、厳重な検査工程を経て、皆様のお手もとに届けられています。

本機を末ながくご愛用いただくために、次の点にご留意のうえ、お取り扱いください。

■ご使用上の注意

- 計算機は精密な電子部品で構成されていますので、絶対に分解しないでください。また、投げたり落したり等のショックや、急激な温度変化を与えないでください。特に、高温の所、湿気やホコリの多い所に放置したり保管することはしないでください。なお、温度が低いときは表示の応答速度が遅くなったり、点灯しなくなることがありますが、通常の温度になると正常にもどります。
- アダプター差し込み口には、本機専用のオプション以外は接続しないでください。
- 計算機の演算中は“—”を表示し、この間のキー操作は一部キーを除いて無効ですから、常に表示を確認しながら、確実にキーを押してください。
- ブザーを鳴らしたときに表示が薄くなることがありますが、故障ではありません。なお、あまり表示が薄いときは、早めに電池を交換してください。
- 電池は、計算機を使わない場合でも2年に1度は交換してください。特に消耗ずみの電池を放置しておきますと、液もれをおこし、故障等の原因になりますので、計算機内には絶対に残しておかないでください。
- 計算機のお手入れは、シンナー・ベンジン等の揮発性液体をさけ、「乾いた柔らかい布」あるいは、「中性洗剤液に浸し固くしぼった布」でおふきください。
- プログラム実行中または演算中には、電源スイッチを切らないでください。

■保証・アフターサービス

- 保証は、別紙の保証書の内容によりますので、よくお読みのうえ、記入事項を確認して、大切に保管してください。
- 万一故障したときは ①お買い上げ店 ②カシオ計算機サービスセンターのうち、ご都合のよい所へ、必ず保証書をそえて、ご持参またはご郵送ください。この場合、故障内容を具体的にお知らせください。
- 修理依頼される前には、この説明書をもう一度お読みになると共に、電源の状態および、プログラムミス、操作ミスがないかをよくお調べください。
- ご不明の点やご質問、お問い合わせ等は、142ページのカシオ計算機へ直接ご連絡ください。

目 次

第1章 覚えておこう本体構成と使い方	1
1-1 各部の名称とそのはたらき	2
1-2 電源について	7
電池交換の仕方	7
オートパワーオフ	8
1-3 計算の前に	9
計算の優先順位	9
入出力桁数と演算桁数	10
第2章 PB-400を動かそう	11
2-1 とにかくさわってみよう	12
2-2 まず始めに基本計算	16
2-3 関数計算もおてのもの	18
第3章 BABICプログラミング	23
3-1 プログラムとは?	24
3-1-1 プログラムはこんなに便利	24
3-1-2 プログラムの仕組み	24
3-1-3 プログラムは簡単だ	26
3-2 プログラムの作成	27
3-2-1 フローチャート(流れ図)を作る	27
3-2-2 プログラムを作る	29
3-2-3 プログラムの入力	32
3-2-4 プログラムの実行	36
3-2-5 デバッグ(まちがいを直す)	39
3-3 発展するプログラム	45
3-3-1 プログラムの流れを変える<GOTO文>	45
3-3-2 プログラムに判断させる<IF~THEN文>	49

目 次

3-3-3	プログラムを繰り返す<FOR・NEXT文>.....	54
3-3-4	複雑なプログラムに便利なサブルーチン <GOSUB・RETURN文>.....	57
3-3-5	関数を使う	61
3-3-6	配列を使う	63
3-3-7	データを読み込む<READ・DATA・RESTORE文>	67
3-3-8	間接指定<ON～GOTO、ON～GOSUB文>	70
3-3-9	文字を扱う文字関数<LEN、MID\$、VAL、STR\$>	72
3-3-10	覚えておくと便利な入出力制御関数<KEY\$、CSR>	74
3-4	あると便利なオプション.....	78
3-4-1	プログラムやデータを保存する.....	78
3-4-2	記録を残す	81
3-5	PB-100のプログラムを使う.....	83

第4章 コマンド・リファレンス

87

NEW{ALL}	89
RUN	89
LIST	90
PASS	91
SAVE{ALL}	92
LOAD{ALL}	93
VERIFY	94
CLEAR	94
END	95
STOP	95
LET	95
REM	96
INPUT	96
KEY\$	97
PRINT	98

CSR	99
GOTO	100
ON~GOTO	101
IF~THEN	102
FOR~NEXT	103
GOSUB	104
RETURN	104
ON~GOSUB	105
DATA	106
READ	107
RESTORE	108
PUT	109
GET	109
BEEP	110
DEFM	111
MODE	112
SET	113
LEN	114
MID\$	115
VAL	116
STR\$	116
SIN、COS、TAN	117
ASN、ACS、ATN	117
LOG、LN	118
EXP	118
SQR	118
ABS	119
SGN	119
INT	119

FRAC	120
RND	120
RAN#	121
DEG	121
DMS\$	122

第5章 ライブラリー

123

■二次回帰分析	124
■万能たてよこ集計	126
■算数博士	131
■ベルトコンベアーゲーム	132
■並びかえゲーム	134

巻末資料

エラーメッセージ一覧表	136
キャラクター表	137
フロチャートの主な記号	138
コマンド索引	140


規 格	141
カシオサービスセンター	142

第 0 章 覚えておこう本体構成と 使い方

今迄にコンピュータに触れたことのない方も、もうすでにコンピュータに慣れた方も、まずこの章はよくお読みください。PB-400の本体構成や使い方を覚えていただくのが、早く使いこなすコツです。

1-1 各部の名称とそのはたらき

まず、PB-400を手にとって、よく見てください。



普通の電卓に比べて、たくさんのキーや差し込み口があります。これだけキーがたくさんあると、どのキーを何に使うのかと迷われるかもしれません。この迷いを取り除くために、これから各々のキーや差し込み口などを説明します。

●電源スイッチ

スライド式のスイッチで、右にスライドすると電源が入り、左にスライドさせると電源が切れます。

●置数キー・小数点キー、計算命令キー、エグゼキュートキー


このキーの配列をよく見てください。普通の電卓と同じ配列をしていますね。この部分はちょうど四則計算(加減乗除)をするときに使いますが、少し異なる点があります。それは **×** (カケル) と **÷** (ワル) のキーがちがっていることと、**=** (イコール) キーがなく、**EXE** (エグゼキュート) というキーがあります。これはコンピュータの言葉では **×** は ***** (アスタリスク) を、**÷** は **/** (スラッシュ) を使い、**=** かわりに **EXE** キーで答えを求めます。




例えば、普通の電卓で $12 \times 4 \div 3 + 7 - 5 =$ と操作するところを、PB-400では、 $12 \times 4 \div 3 + 7 - 5 \text{ EXE}$ と操作します。

これで、PB-400を普通の電卓がわりに使うこともできます。

なお、**[SHIFT]**キー(赤色の**[S]**キーをアルファベットキーの**[S]**と区別するために本書では**[SHIFT]**と記します。)に続けて押しますと、**[P0]~[P9]**キーは**P0~P9**のプログラムエリア指定の役目をし、**[x^y]**キーはべき乗計算($x^y \rightarrow x \uparrow y$)の、**[+][−][*][÷]**キーは大小比較記号($\geq, \leq, >, <$)を表示します。




●アルファベットキー、スペースキー



このキーがPB-400の特長で、タイプライターのようにアルファベット26文字と、スペースキー(**[SPC]**)が並んでいます。このキーを使ってPB-400に命令を与えたり、プログラムを書き込んだりします。また、A~Zまでの26文字のキーはそれぞれがメモリー(記憶するところ)の役をします。


なお、**[A]~[Z]**のキーには別の役割があり、**[SHIFT]**(赤い**[S]**のキーで、アルファベットキーの**[S]**と区別するために本書ではこのように記します。)に続けて押すと記号やBASICのコマンドを表示します。

例) **[SHIFT][A] → GOSUB**、**[SHIFT][U] → ?**



このアルファベットキーはあと2つの顔を持っています。それは拡張モード (MODEキーに続いて□キーを押す。EXT点灯)での使い方で、直接押すとアルファベットの小文字を、SHIFTキーに続けて押すと特殊記号を表示します。

拡張モードでの働き



拡張モードでSHIFTキーに続けて押したときの働き



拡張モードを解除して、アルファベット大文字に戻すには、もう一度MODE□と押します。

このように、アルファベットキーは4つの顔を持っていますので、よく覚えておいてください。

●イコールキー(=)

このキーは計算の答を求めるためのキーではなく、代入文(28ページ参照)やIF文(49ページ参照)での判断のために使います。

なお、SHIFTキーに続いて押しますと、≠(等しくない)の記号が表示されます。

●指数部置数キー・パイキー(⌘)




このキーは直接押すと指数部置数キーとなり、指数部(10の何乗)を置数する前に押します。例えば 1.23×10^4 の場合は1□2□3□E□4と押します。指数部が負の場合は、このキーに続いて-□キーを押します。 7.41×10^{-9} は7□4□1□E□-□9と押します。


SHIFTキーに続いて押しますと π (パイ……円周率)を表示します。


●アンサーキー(ANS)


このキーは直前に行なわれた計算の答を覚えているキーで、マニュアル計算とプログラム計算で行なわれた計算の答を表示します。


●モードキー (MODE)


このキーは計算機の状態や角度単位を指定するするときに、  ~  のキーと組み合わせて使います。


 "EXT"を表示し、拡張モードとなり、英小文字・特殊記号が使えます。再び押すと拡張モードを解除します。


 "RUN"を表示し、マニュアル計算およびプログラム計算が行なえます。

 "WRT"を表示し、プログラムの書き込みおよびチェック、編集が行なえます。


 "TR"を表示し、実行トレースが行なえます。(詳しくは44ページ)


 "TR"が表示している場合は"TR"が消え、実行トレース機能が解除されます。

 "DEG"を表示し、角度の単位を<度>に指定します。

 "RAD"を表示し、角度の単位を<ラジアン>に指定します。

 "GRA"を表示し、角度の単位を<グラジアン>に指定します。

 "PRT"を表示し、プリンタ接続時はプリント出力をすることができます。

 "PRT"が表示している場合は"PRT"が消え、プリント出力が解除されます。

●カーソルキー (←→)


このキーは表示されている文字を訂正するときに便利なキーで、カーソル(表示窓で点滅している"_"のことです)を左右に移動させます。1回押すと一文字分移動し、押し続けると文字の書いてある範囲内を続けて移動します。

●オールクリアーキー (AC)

このキーは全てのキーの中で一番強いキーで、どんな表示でも消してしまいます。また、エラーになって停止したときやオートパワーオフ(8ページ参照)で表示が消えているときにも押します。プログラム実行中は、プログラムを中断させます。

●インサート・デリートキー (INS)

このキーも表示されている文字を訂正するときに便利なキーで、カーソルが点滅している文字を削除(デリート)します。削除した後はカーソルより右側の文字を左につめます。


また、キーに続けて押しますと、カーソルの点滅している文字以降を右にずらし、空白をあけます。

● ストップキー (STOP)

このキーはプログラム実行中に使うキーで、プログラムの実行を一時的に停止させます。プログラムを続けて実行させたいときは**EXE**キーを押すことにより再開します。

● 表示用コントラスト

電池の消耗や表示窓を見る角度により、濃く見えたり薄く見えたりします。このようなときには、本体右側面にあるボリュームで、見やすい濃さに調整してください。



ボリュームは矢印方向に回すと濃くなり、逆に回すと薄くなります。

なお、最も濃い位置にしてもまだ表示が薄い場合は、電池がかなり消耗していることが考えられますので、なるべく早く電池を交換してください。

● アダプター差し込み口

この差し込み口は別売のオプションを接続するコネクタで、プリンタを使うときは<FP-12>を、テープレコーダーで記録するときには<FA-3>をつなぎます。



この差し込み口には<FP-12>、<FA-3>以外は接続しないでください。

また、オプションを接続しないときは、必ず付属のコネクタカバーをつけて使用してください。

1-2 電源について


本機は、リチウム電池(CR-2032)2個を電源とします。

コントラストのボリューム(6ページ参照)を最も濃くなる位置にしても表示が薄く見えるときは電池が消耗していますので、早目に下記に示すように電池交換をしてください。


なお、計算機が正常に使用できても、2年に1度は必ず電池交換をしてください。

■電池交換の仕方

- (1)電源スイッチを切ってから、裏面の2個のネジをはずし、裏ブタをはずします。




- (2)図の①を押しながら矢印方向にスライドさせて、電池押さえ板をはずします。



- (3)古い電池を2個とも取り出します。

(電池ボックスを下に向けて軽くたたけば簡単にはずれます。)

- (4)新しい電池の表面を乾いた布でよくふいてから⊕側を上にして入れます。



- (5)電池押さえ板で電池を押さえながらスライドさせてとめます。

- (6)裏ブタをネジ止めし、電源スイッチをONにしてからリセットボタンを先の細い棒で押してください。

※ブザーを多用しますと電池寿命は短くなります。

※電池は必ず2個とも交換してください。

※消耗済みの電池は絶対に火中に投入しないでください。破裂することがあり、非常に危険です。

※電池の⊕⊖は絶対にまちがえないようにしてください。

電池は、幼児の手のとどかないところに保管してください。

万一、飲み込んだ場合にはただちに医師と相談してください。

■オートパワーオフ(自動電源OFF)

スイッチの切り忘れによるムダな電力消費を防ぐ自動節電機能で、操作完了後(プログラム計算中を除く)約7分で自動的に電源オフになります。

この場合は、電源スイッチを入れ直するか、**AC**キーを押せば、再び電源オンになります。

※電源オフになってもメモリー内容およびプログラム内容は消えませんが、角度 指定や各モード指定(“WRT”、“TR”、“PRT”等)はすべて解除されます。

1-3 計算の前に

■計算の優先順位


計算には「優先順位」という規則があり、たし算・ひき算よりかけ算・わり算の方を先に計算することになっています。PB-400はこの優先順位を計算機自身が自動的に判別するようにできています。この機能はとても便利で、数式をそのまま打ち込んでいけば正しい答が求められます。

計算の優先順位は次のように定められています。

- ①関数 (SIN, COS等)
- ②べき乗 (↑)
- ③×(*), ÷(/)
- ④+, -

計算はこの優先順位に従って行なわれますが、優先順位が同じときは頭から(左から)、カッコが使用されたときはカッコの内が最優先されます。

〈例〉 $2 + 3 * \text{SIN} (17 + 13) \uparrow 2 = 2.75$



■入出力桁数と演算桁数

本機の数値を入力できる範囲(入力桁数)は仮数部12桁、指数部2桁です。内部演算も同じで、仮数部12桁、指数部2桁で行なわれます。

数値を表示できる範囲(出力桁数)は通常、仮数部10桁で、マニュアル計算による答の表示とプログラム中での答の表示は異なります。マニュアル計算では指数部や負符号がつく場合、仮数部、指数部、負符号を含めて12桁まで表示します。プログラム中では仮数部10桁、指数部2桁を全て表示しますが、合計が12桁をこえる場合、最初に頭から12桁分を表示し、以後順に表示が左に送られるようにして表示されます。

〈例〉

マニュアル計算

1□2345678912 **EXE**
 12345678912 **✖** 100 **EXE**
 12345678912 **✖** -100 **EXE**

1.234567891
1.2345678 E12
-1.234567 E12

プログラム計算

PRINT 12345678912 **✖** -100 の場合

	-1.234567891	E11
-	1.234567891E	11
-1	.234567891 E1	1
-1.	234567891 E11	

表示の外に消えます。

まだ表示されていません。

自動的に送られます。

第2章


PB-400を動かそう

ここでは、PB-400に慣れ親んでいただくために、とにかくさわってください。多少まちがった操作をしても壊れるものではありません。習うより慣れろのことわざ通りに、まずは簡単な操作から。


2-1 とにかくさわってみよう




ここではまずさわって動かしてみて、どのように動くかを見てみましょう。
まず最初にするのは、PB-400を手にとって、電源スイッチを右にスライドさせて、電源を入れます。





すると表示は次のようになるはずです。



初めにこの表示を消してみます。**AC**キーを押してください。“READY P0”は消えましたね。このとき左端に点滅している“-”があります。これを「カーソル」と呼び、ここから文字を書き始めます。



このカーソルが点滅している状態を「入力待ち状態」とも言い、計算や命令を待っているのです。カーソルは通常“-”の点滅ですが、文字を続けて書いていくうちに、“”の点滅となることがあります。PB-400では1行に書ける文字数が62文字までですので、56文字以上書きますと注意信号として表われます。なお、表示窓の上に“RUN”と“DEG”が点灯していると思いますが、これは状態表示といい、今PB-400がどのような状態になっているかを示します。“RUN”はRUN(ラン)モードを示し、マニュアル計算やプログラムの実行が行なえます。“DEG”は角度単位で、度になっていることを示します。角度単位はほかに、**MODE** と押して指定するラジアンモード(“RAD”点灯)、**MODE** と押して指定するグラジアンモード(“GRA”点灯)があります。この角度単位は三角関数などを使うときに必要となります。電源スイッチONでは“DEG”が表示されます。

状態表示はほかにも**MODE** と押すプログラム書き込みモード(“WRT”点灯)、**MODE** と押すトレースモード(“TR”点灯、44ページ参照)、**MODE** と押すプリントモード(“PRT”点灯、81ページ参照)、**MODE** と押す拡張モード(“EXT”点灯)を示すものがあります。

このような状態表示はさわっているうちに覚えてきますので、ここでは見るだけにしてもかまいません。

では、計算機にさわって表示をさせてみましょう。

もしいろいろさわって、状態表示が色々と点灯している方は、一度電源スイッチを切ってから再び電源を入れてください。

まず初めに簡単な計算をしてみましょう。

例) $123+456=579$

ACを押します。

数式の通りにキーを押します。

1 2 3 + 4 5 6

この後に **=** のかわりに **EXE** で答を求めます。

EXE

—

123+456—

579

どうでしたか、計算が簡単にできますね。

では、次にもう少し長い計算をしてみましょう。

例) $45 \times 6 + 89 = 359$

ここでは45をまちがえて46と押してしま
ったとします。

4 6 * 6 + 8 9

45を46と押してしまったことに気がつき
ました。でも、あわてずに、カーソルキ
ー(**←**)でカーソルをまちがった所に合わ
せます。

← ← ← ← ← ←

ここで正しい**5**のキーを押します。

5

これで計算式が正しくなりましたので、
答を求めます。

EXE

46*6+89—

4~~6~~*6+89

カーソルと6とが交互に
点滅します

45*6+89

359

このように、途中でまちがいに気付いたときは、カーソルキーを使って簡単に
訂正することができます。ただし、**EXE**キーを押してしまった後では、最初から
入れなおしてください。

それでは、アルファベットキーを使って文字を書いてみましょう。

アルファベットキーはタイプライターと同じ配列になっており、この配列をASCII型配列といいます。現在のパソコンはこのASCII型配列のキーボードが主流ですので、最初は慣れないでしょうが、だいたい位置は覚えておいてください。

まず、計算機の名前を書いてみましょう。

例) 名前は「CASIO PB-400」とします。

CASIOと書いてみます。

C A S I O

CASIO

次にPB-400と書いてみます。

P B - 4 0 0

CASIOPB-400

それでは、CASIOとPB-400の間を1文字分あけてみます。

カーソルをPに合わせます。

← ← ← ← ← ←

CASIOPB-400

1文字分あけます。

SHIFT INS

CASIO_PB-400

文字間をあけるときは、あけたい箇所の次にカーソルを合わせ、SHIFT INSで1文字分あきますので、何文字分もあけたいときは、繰り返し同じ操作をします。

PB-400には数字・アルファベットのほかに、いくつかの特殊文字と呼ばれる文字があります。これはゲームや科学記号に便利な文字です。特殊文字の種類については4ページを参照してください。

ここで少し表示させてみましょう。

例) ♠♥♦♣のマークを表示させる。

まず拡張モードを指定します。

AC MODE

点灯します

EXT

このマークはSHIFTキーに続けてアルファベットキーを押します。

SHIFT ♠ SHIFT ♥ SHIFT ♦ SHIFT ♣

♠♥♦♣


例) ΣΩμの記号を表示させる。

拡張モードになっているので、そのまま

SHIFTキーに続けて押します。

SHIFT Σ SHIFT Ω SHIFT μ

♠♥♦♣ΣΩμ

このようなマークや記号が用意されていますので、色々と利用して使ってみてください。なお、拡張モードになっているときに、今迄のアルファベット大文字を表示するモードに戻す場合は、もう一度 **MODE**  と押し、**"EXT"**を消します。これでキーの押し方はだいたいわかったかと思います。このように色々とさわっているうちに**"ERR 2"**が表示されて、キーを押しても動かなくなることがあります。これは故障ではなく、PB-400にまちがった命令をしましたというメッセージで、「エラーメッセージ」と呼ばれるものです。このときはあわてずに、**AC** キーを押せば表示が消えて、また動くようになります。このようなエラーメッセージにはいくつかありますので、詳しくは 39ページまたは136ページをご覧ください。

2-2 まず始めに基本計算

ここでは簡単な四則計算ぐらいを行なってみますが、関数電卓を使ったことのない方は注意していただきたいのです。それは、PB-400も完全数式通りというたし算、ひき算よりかけ算、わり算を先に計算する機能を持っているからです。

例1) $23+4.5-53=-25.5$

操作) $23 \text{ + } 4.5 \text{ - } 53 \text{ EXE}$

-25.5

※ここからは、数字キーはワクをはずして記します。

例2) $56 \times (-12) \div (-2.5) = 268.8$

操作) $56 \text{ * } 12 \text{ / } 2.5 \text{ EXE}$

268.8

※負符号がつく場合は、数字の前に +/- キーを押します。

例3) $7 \times 8 - 4 \times 5 = 36$

操作) $7 \text{ * } 8 \text{ - } 4 \text{ * } 5 \text{ EXE}$

36

※かけ算を先に計算してから、ひき算を計算します。

例4) $(4.5 \times 10^{75}) \times (-2.3 \times 10^{-78}) = -0.01035$

操作) $4.5 \text{ E } 75 \text{ * } 2.3 \text{ E } 78 \text{ EXE}$

-0.01035

※指数部は E キーに続いて押します。

もう一つの計算として、メモリーを使った代数計算があります。この計算は、ある一定の数値を色々と計算するときに便利です。

例えば $3x + 5 =$

$4x + 6 =$

$5x + 7 =$

というような計算があり、 x の値が123.456であるとき、同じ数値を繰り返し押すのは面倒なものです。この計算を手間をかけずに行なう方法はないでしょうか。解決策は変数と呼ばれるメモリーを使うことです。この例では代数計算に x という代数を使っていますので、変数 X を使って計算します。

まず、変数Xに123.456を入れます。

X **=** 123.456 **EXE**

この**=**は等しいという意味ではなく、変数Xに123.456を入れるという意味です。
では、計算をしてみましょう。

3 **×** **X** **+** 5 **EXE**

4 **×** **X** **+** 6 **EXE**

5 **×** **X** **+** 7 **EXE**

375.368
499.824
624.28

こんなに簡単にできます。

PB-400にはこのような変数がA～Zまで26個ありますので、色々な数値を覚えておくことができます。

この例では変数Xの数値は一定で、計算式が異なりますが、逆に計算式が一定で、変数の値が異なる場合はどうでしょう。

もし、計算式が $3x + 5 =$ と決っていて、 x の値が123.456.789と変化する場合、今のような方法では操作が面倒になります。実際には計算式をPB-400が覚えてくれて、変数Xの値だけを変えればよいのです。この便利な計算方法を「プログラム計算」といいます。PB-400はこのプログラム計算が得意なのです。ここではプログラムを使う前のマニュアル計算を行なっていますので、プログラムについては、第3章のプログラム編をご覧ください。

2-3 関数計算もおてのもの

PB-400は一般の四則計算のほかに、関数計算の機能も持っています。
この関数機能はプログラム中に組み込んでも使えますが、ここではマニュアルでの使い方について説明します。

PB-400に組み込まれている関数は次の通りです。

関数名	書式
三角関数	
$\sin x$	SIN x
$\cos x$	COS x
$\tan x$	TAN x
逆三角関数	
$\sin^{-1} x$	ASN x
$\cos^{-1} x$	ACS x
$\tan^{-1} x$	ATN x
平方根	\sqrt{x} SQR x
常用対数	$\log x$ LOG x
自然対数	$\ln x$ LN x
指数関数	e^x EXP x
べき乗	x^y $x \uparrow y$
10進→60進	DMS \$(x)^*\$
60進→10進	DEG(x, y, z) [*]
整数化	INT x
整数部除去	FRAC x
絶対値化	$ x $ ABS x
符号化	$\begin{cases} \text{正数} \rightarrow 1 \\ 0 \rightarrow 0 \\ \text{負数} \rightarrow -1 \end{cases}$ SGN x
四捨五入	$\begin{cases} x \text{ の } 10^y \text{ を} \\ \text{四捨五入} \end{cases}$ RND(x, y) [*]
乱数	RAN#

※ DMS \$、DEG、RND は引数を必ず()でくくります。

では、関数を使って計算をしてみましょう。

●三角関数(sin、cos、tan)、逆三角関数(\sin^{-1} 、 \cos^{-1} 、 \tan^{-1})

三角・逆三角関数を使うときは、必ず角度単位の指定(DEG、RAD、GRA)を行なってください。(角度単位を変更しない場合は、新たに行なう必要はありません。)

<例> $\sin 12.3456^\circ = 0.2138079201$

<操作> MODE $\boxed{4}$ \rightarrow "DEG"

SIN 12.3456 EXE

0.2138079201

※ここからは、アルファベットキー、数字キーはワクをはずして記します。

<例> $\cos 63^\circ 52' 41'' = 0.4402830847$

<操作> COS DEG $\boxed{\text{SHIFT}}$ $\boxed{\text{6}}$ 63 $\boxed{\text{SHIFT}}$ $\boxed{\text{5}}$ 52 $\boxed{\text{SHIFT}}$ $\boxed{\text{4}}$ 41 $\boxed{\text{SHIFT}}$ $\boxed{\text{EXE}}$

0.4402830847

<例> $2 \cdot \sin 45^\circ \times \cos 65.1^\circ = 0.5954345575$

<操作> 2 * SIN 45 * COS 65.1 EXE

0.5954345575

<例> $\sin^{-1} 0.5 = 30^\circ$

<操作> ASN 0.5 EXE

30

<例> $\cos \left(\frac{\pi}{3} \text{rad} \right) = 0.5$

<操作> MODE $\boxed{5}$ \rightarrow "RAD"

COS $\boxed{\text{SHIFT}}$ $\boxed{\text{1}}$ $\boxed{\text{SHIFT}}$ $\boxed{\pi}$ 3 $\boxed{\text{SHIFT}}$ $\boxed{\text{EXE}}$

0.5

<例> $\cos^{-1} \frac{\sqrt{2}}{2} = 0.7853981634 \text{rad}$

<操作> ACS $\boxed{\text{SHIFT}}$ $\boxed{\text{1}}$ SQR 2 $\boxed{\text{2}}$ 2 $\boxed{\text{SHIFT}}$ $\boxed{\text{EXE}}$

0.7853981634

<例> $\tan(-35\text{gra}) = -0.612800788$

<操作> MODE $\boxed{6}$ \rightarrow "GRA"

TAN $\boxed{-}$ 35 EXE

-0.612800788

●対数関数(log、ln)、指数関数(e 、 x^y)

<例> $\log 1.23 (= \log^{10} 1.23) = 0.0899051114$

<操作> LOG 1.23 EXE

0.0899051114

〈例〉 $\ln 90 (= \log_e 90) = 4.4980967$

〈操作〉 LN 90 **EXE**

4.4980967

〈例〉 $e^5 = 148.4131591$

〈操作〉 EXP 5 **EXE**

148.4131591

〈例〉 $10^{1.23} = 16.98243652$

(常用対数1.23の真数を求める)

〈操作〉 10 **SHIFT** \square 1.23 **EXE**

16.98243652

〈例〉 $5.6^{2.3} = 52.58143837$

〈操作〉 5.6 **SHIFT** \square 2.3 **EXE**

52.58143837

〈例〉 $123^{\frac{1}{7}} (= \sqrt[7]{123}) = 1.988647795$

〈操作〉 123 **SHIFT** \square **SHIFT** \square 1 **7** **SHIFT** \square **EXE**

1.988647795

〈例〉 $\log \sin 40^\circ + \log \cos 35^\circ = -0.278567983$

その真数は……0.5265407845 ($\sin 40^\circ \times \cos 35^\circ$ の対数計算)

〈操作〉 **MODE** **4** \rightarrow "DEG"

LOG SIN 40 **+** LOG COS 35 **EXE**

10 **SHIFT** \square **ANS** **EXE**

-0.278567983

0.5265407845

● その他の関数 ($\sqrt{\quad}$ 、SGN、RAN $\#$ 、RND、ABS、INT、FRAC)

〈例〉 $\sqrt{2} + \sqrt{5} = 3.65028154$

〈操作〉 SQR 2 **+** SQR 5 **EXE**

3.65028154

〈例〉 正数であれば"1"を、負数であれば"-1"を、0であれば"0"を与える。

〈操作〉 SGN 6 **EXE**

SGN 0 **EXE**

SGN **-** 2 **EXE**

1

0

-1

〈例〉 乱数発生 ($0 < \text{RAN}\# < 1$ の擬似乱数)

〈操作〉 RAN **SHIFT** \square **EXE**

0.7903739076

〈例〉 12.3×4.56 の答を 10^{-2} の位で四捨五入する。

$$12.3 \times 4.56 = 56.088$$

〈操作〉 $\text{RND} \left[\text{SHIFT} \right] \left[\text{1/x} \right] 12.3 \left[\text{SHIFT} \right] \left[\text{1/x} \right] 4.56 \left[\text{SHIFT} \right] \left[\text{1/x} \right] 2 \left[\text{SHIFT} \right] \left[\text{1/x} \right] \text{EXE}$

※ $\text{RND}(x, y)$ のときの y は $|y| < 100$

56.1

〈例〉 $|-78.9 \div 5.6| = 14.08928571$

〈操作〉 $\text{ABS} \left[\text{SHIFT} \right] \left[\text{1/x} \right] 78.9 \left[\text{SHIFT} \right] \left[\text{1/x} \right] 5.6 \left[\text{SHIFT} \right] \left[\text{1/x} \right] \text{EXE}$

14.08928571

〈例〉 $\frac{7800}{96}$ の整数部は……81

〈操作〉 $\text{INT} \left[\text{SHIFT} \right] \left[\text{1/x} \right] 7800 \left[\text{SHIFT} \right] \left[\text{1/x} \right] 96 \left[\text{SHIFT} \right] \left[\text{1/x} \right] \text{EXE}$

81

※ この関数は元の数値をこえない最大の整数を求めます。

〈例〉 $\frac{7800}{96}$ の小数部は……0.25

〈操作〉 $\text{FRAC} \left[\text{SHIFT} \right] \left[\text{1/x} \right] 7800 \left[\text{SHIFT} \right] \left[\text{1/x} \right] 96 \left[\text{SHIFT} \right] \left[\text{1/x} \right] \text{EXE}$

0.25

●有効桁数指定、小数以下指定

有効桁数と小数以下の指定は“SET”コマンドにより行ないます。

有効桁数指定……SET E n ($n = 0 \sim 8$)

小数以下指定……SET F n ($n = 0 \sim 9$)

指定解除……SET N

※ マニュアル計算での有効桁数指定の場合の“SET E 0”は8桁指定となります。

※ 指定を行なうと、指定桁の下1桁目を四捨五入して表示します。

なお、計算機内部やメモリー内にはもとの数値が残っています。

〈例〉 $100 \div 6 = 16.66666666 \dots$

〈操作〉 SET E 4 EXE (有効桁数4桁指定)

100 $\left[\text{SHIFT} \right] \left[\text{1/x} \right] 6 \left[\text{SHIFT} \right] \left[\text{1/x} \right] \text{EXE}$

1.667 E01

〈例〉 $123 \div 7 = 17.57142857 \dots$

〈操作〉 SET F 2 EXE (小数以下2桁指定)

123 $\left[\text{SHIFT} \right] \left[\text{1/x} \right] 7 \left[\text{SHIFT} \right] \left[\text{1/x} \right] \text{EXE}$

17.57

〈例〉 $1 \div 3 = 0.33333333 \dots$

〈操作〉 SET N EXE (指定解除)

1 $\left[\text{SHIFT} \right] \left[\text{1/x} \right] 3 \left[\text{SHIFT} \right] \left[\text{1/x} \right] \text{EXE}$

0.33333333

● 10進↔60進変換(DEG、DMS \$)

〈例〉 $14^{\circ}25'36'' = 14.42666667$

〈操作〉 DEG $\boxed{\text{SHIFT}} \boxed{\text{↵}} 14 \boxed{\text{SHIFT}} \boxed{\text{↵}} 25 \boxed{\text{SHIFT}} \boxed{\text{↵}} 36 \boxed{\text{SHIFT}} \boxed{\text{↵}} \boxed{\text{EXE}}$

14.42666667

〈例〉 $12.3456^{\circ} = 12^{\circ}20'44.16''$

〈操作〉 DMS $\boxed{\text{SHIFT}} \boxed{\text{↵}} \boxed{\text{SHIFT}} \boxed{\text{↵}} 12.3456 \boxed{\text{SHIFT}} \boxed{\text{↵}} \boxed{\text{EXE}}$

12° 20' 44.16

〈例〉 $\sin 63^{\circ}52'41'' = 0.897859012$

〈操作〉 $\boxed{\text{MODE}} \boxed{4}$

SIN DEG $\boxed{\text{SHIFT}} \boxed{\text{↵}} 63 \boxed{\text{SHIFT}} \boxed{\text{↵}} 52 \boxed{\text{SHIFT}} \boxed{\text{↵}} 41 \boxed{\text{SHIFT}} \boxed{\text{↵}} \boxed{\text{EXE}}$

0.897859012

関数計算もこのように簡単にできます。

第 3 章 BASIC プログラミング

本章では、主に BASIC のプログラムとはどのようなことか、どのように作るのかについて説明してあります。今迄、プログラムに慣れていない方は、この章を繰り返しお読みになって、基本をマスターしてください。

3-1 プログラムとは？

「プログラム」と聞くとよく難しいものだと感じる方がいますが、プログラムにも簡単なものから難しいものまであり、代数式を記憶させ、その代数式に数値を代入して計算させるのも立派なプログラムです。

まずは難しい考え方は抜きにして、楽な気持ちで話を進めていきましょう。

3-1-1 プログラムはこんなに便利

私達の身のまわりには、色々な計算があります。仕事で使う会計や金融計算、測量や計測、その他にも家計簿や経費の計算などと考えられるだけでも多いものです。この計算も1回だけで終るものならかまいませんが、何度も繰り返し同じ計算式で数値を変えて計算するのも大変な手間です。こんな計算は、PB-400にとって最も得意な仕事です。たとえば、 $y=2x^2+5x+13$ という数式があるとします。この中で x の値が変化するときの y の値を求めるとすれば、 x の値をかえながら同じ計算をしなければなりません。この手間をはぶくために、この式を記憶させてみましょう。

まずは見てください。

```
10 INPUT X
20 Y=2*X↑2+5*X+13
30 PRINT Y
```

これが数式を記憶させたプログラムです。

細かい説明は後にするとして、2行目は計算式をそのまま書いています。これでも立派なプログラムで、計算が便利になります。

このようにプログラムとは、けっして難かしいものばかりではなく、身近な計算式をそのまま記憶させるだけでもかなり便利に使えます。

それでは、プログラムについて順に説明していきましょう。

3-1-2 プログラムの仕組み

まず、プログラムの仕組みを覚えてください。

```
10 INPUT X
20 Y=2*X↑2+5*X+13
30 PRINT Y
```

これは前に出てきたプログラムです。このプログラムを分解して仕組みを見てみましょう。

このプログラムは大きく分けると3つに分けられます。

```
10 INPUT X .....入力部
20 Y=2*X↑2+5*X+13 .....計算部
30 PRINT Y .....出力部
```

最初の入力部はデータ(計算に必要な数値など)を計算機に入れる(入力する)部分で、データの数により増えます。2つ目の計算部は計算をさせて答を求める部分で、一番の要となります。最後の出力部は計算機に答えてもらう(表示させる)部分で、計算をただけでは何も答をおしえてくれませんので、答を表示させます。

計算機はなんでもしてくれますが、正しい命令を与えてやらなければ何もしてくれません。そのために、**データを入れなさい(入力部)**という命令と、**答を表示しなさい(出力部)**という命令を記憶させるのです。

この3つの部分をさらに細かく分解すると、次のようになります。

<u>10</u>	<u>INPUT</u>	<u>X</u>
行番号	コマンド	オペランド

行番号とは、プログラムの流れにそってつける順番を示すもので、この行番号の小さい方から順に計算機が読んで、実行していきますから、実行させたい順につけてください。なお、この例のように、10、20、30……と10刻みに書いているのは、後で追加が必要となったときに便利にするためです。本当は1、2、3……と続けてもよいのですが、1.5とか、12.3のように小数点以下はつけられません。

行番号の次に続くのが**コマンド**で、計算機にどのようなことをさせるかの命令を与えます。このコマンドは色々あり、命令により使い分けます。本当は全部のコマンドを覚えていただきたいのですが、一度に覚えるのは大変ですので、最低限必要なものを覚えていただいて、少しづつ数を増していってください。コマンドの種類や機能については、87ページからの「コマンドリファレンス」を参照してください。

コマンドの後の**オペランド**は、コマンドの持つ命令を補足するためにあり、つくものとつかないものがあります。この例ではコマンドが"INPUT"ですので、データを入力しなさいという意味があり、その入力したデータを入れるメモリーを指定するのがオペランドで、この場合変数Xに入れなさいという意味になります。

次の行では、

<u>20</u>	<u>Y=2*X↑2+5*X+13</u>
行番号	代入文

行番号は前に続けて20としています。次の代入文は=(イコール)の右辺の値を左辺の変数に入れる(代入する)という意味です。この代入文も実際はコマンドとして"LET"がつき、

20 LET Y=2*X↑2+5*X+13

とすれば"LET"がコマンドで、代入文がオペランドということになります。行番号30は、行番号10と同じで、行番号、コマンド、オペランドでできています。

<u>30</u>	<u>PRINT</u>	<u>Y</u>
行番号	コマンド	オペランド

プログラムとは、だいたいこのような仕組みでできており、これ等のコマンドやオペランドが多くなったり、行数が増えたりして大きなプログラムとなります。

3-1-3 プログラムは簡単だ

プログラムの仕組みさえ知ってしまえば、プログラムなんてたいしたものではないことがわかりだと思えます。

プログラムを作るには、入力部、計算部、出力部の三本の柱がわかれば十分に活用できます。全部のコマンドを一度に覚える必要はないのです。まずは簡単な"INPUT"、"PRINT"、それに計算の代入文を使って、身近な計算をやらせてみるのです。

プログラムを早く覚えるコツは、ただ漠然とプログラムを覚えようとせずに、身近な問題を探してプログラムにしてみるのです。会社で繰り返し行なう伝票計算や金種計算、測量や設計の計算、家での家計簿やオーディオの時間計算など、考えてみれば色々あるはずです。この問題の中で計算式にしやすいものを探して、プログラム化してみるのです。ただ漠然と覚えようとせずに、1つの目的を持ってそのプログラムを作り、それに必要なコマンドをまず覚えるのです。あとは、この作業を繰り返しながらプログラムをより便利なものへと改良していけばよいのです。

もう一つのコツは、全体を一度に作ろうとはせずに、部分的に作り、あとで一つにするのです。どんなに大きくて複雑なプログラムも、部分的に分ければより分かりやすくなるはずですよ。

これ等のことを考えながら、少しずつでいいですから、ゆっくりと行なってください。基本さえマスターすれば、後はコマンドリファレンス編で各コマンドの機能を読みながら使うだけで立派にプログラムが作れます。

それでは、プログラムを作ってみましょう。


3-2 プログラムの作成

ここからがBASICプログラミングの本題で、実際にプログラムを作り上げます。


3-2-1 フローチャート(流れ図)を作る


フローチャートという名前はあまりなじみのない名前かもしれませんが。これは作業の順番を図式化したもので、流れ図ともいいます。

よく「BASICにはフローチャートはいらない」という言葉を聞きますが、それは頭の中でフローチャートが描ける人が言う言葉で、慣れないうちは全体の作業の段取りがわからないものです。そのために、簡単なフローチャートを描いて流れをつかむことが大切です。

フローチャートを描くにも正式には専門の記号がありますが、そんな記号は覚えずに、1つの作業ごとに  で囲んで線で継ぐだけでよいのです。


では実際にプログラムを作りながら説明していきましょう。

例として、ある数を入力して、その数の2乗を求めるプログラムを作ってみましょう。まず、部分的に考えますと、計算部があります。そして、ある数を入力するので入力部があります。それに、答を表示させるわけですから出力部があります。この3つを  で囲むと次のようになります。




この3つの要素を順に継げるわけで、最後にくるのは答の表示だということがすぐにわかると思います。次に、答を求めるために計算をしなければなりません。計算をするためには、データを入力してやらねばなりません。

これで3つの順番が分ってきたと思います。では、この3つを続けてみましょう。



これでフローチャートはでき上がりましたが、このフローチャートをさらにプログラムに近い形にしてみましょう。

1 番目はデータを入力する命令です。これはINPUT文を使い、変数にデータを
 入力させますので、変数を決めます。ここでは、仮に変数Aを使いますので、
 ①の内容は、" INPUT A"となります。2 番目の計算は、入力された変数Aの
 内容を2乗し、答を別の変数に入れます。もう一つの変数をBとしますと、
 "B=A↑2"となります。この計算式は代入文と言い正式にば"LET B=A↑2"
 と書きますが、LETは省略できますので"B=A↑2"だけでもかまいません。そ
 して、3 番目では答を表示させます。PRINT文を使い、答の入っている変数Bの
 内容を表示させますので、"PRINT B"となります。この3つの作業をもう1
 度フローチャートにします。



この後は、3つの作業に行番号をつければプログラムのでき上りです。


```

10 INPUT A
20 B=A↑2
30 PRINT B
  
```

このように、フローチャートを作りながら順に組み立てていった方が、プログ
 ラムを早く、簡単に作るができます。

通常は、この2つのフローチャートの内の1つだけ作ればよいのですが、慣れ
 ないうちは1つ目のフローチャートを作るようにした方が便利です。

実際のフローチャートには正式な記号があり、それぞれ意味があります。この
 記号についてはあまり意識しなくてもかまいませんが、この例を正式な記号で
 描いてみましょう。




これ等の記号は巻末に記載してありますので、参照してください。

3-2-2 プログラムを作る

これまでは、プログラムを作る前の段取りを話してきましたが、ここからは実際にプログラムを作ってみましょう。

まず簡単な例として、2つの数値を入れて、その2つの数値の和・差・積・商を求めてみましょう。

ここでも重要なものは3本の柱である入力部、計算部、出力部で、この考え方に当てはめフローチャートを作ってみましょう。



最初に2つの数値を入力するわけですから、INPUT文を使います。INPUT文はキーボードからデータを変数に入力する命令です。ここで1つ考えることは、データや答を入れておく変数です。プログラムを作っていく上で迷いがちなのが変数の使い方です。この変数はAからZまでのアルファベット1文字の名前を持つものと、A(3)などのように添字とよばれる要素を伴う配列変数があります。これ等の変数の中から選び出すわけですが、慣れないうちに、出てくる順番にA、B、C、……と選べばよいと思います。

ここでは2つの数値を入れますので、AとBを選び、“INPUT A,B”とします。このINPUT文はカンマ(,)で区切ることにより、1つでいくつもの変数を扱えます。

それでは、次の計算部はどうでしょうか。ここでは4つの計算を行なうわけですから、答も4つ出てきます。この4つの答を入れる変数をここではC、D、E、Fとします。

まず加算は、 $A+B$ ですので、 $C=A+B$ とします。

減算は、 $A-B$ ですので、 $D=A-B$ とします。


積算は、 $A*B$ ですので、 $E=A*B$ とします。

除算は、 A/B ですので、 $F=A/B$ とします。

これで、計算部はできました。次は、この答を表示させます。

表示させる命令はPRINTですから、“PRINT C,D,E,F”としましょう。

このプログラム化した型式を再びフローチャートにしてみますと、次のようになります。



これに行番号をつけてプログラムを完成させます。

```
10 INPUT A,B
20 C=A+B
30 D=A-B
40 E=A*B
50 F=A/B
60 PRINT C,D,E,F
```

そして、最後にプログラムの終了を示す“END”をつけます。


```
70 END
```

これがプログラムが完成しました。このように順をおって組み立てていくと、プログラムも簡単に作れます。最初から色々な命令を使ってむずかしく考えるよりも、簡単でも使えるプログラムを作りましょう。

ワンポイントレッスン


変数とは

プログラムを作る上で重要な要素として、変数があります。変数とは、入力したデータや計算の答を入れておく箱で、各々に名前がついています。この変数には標準のAからZまでの変数と、配列変数と呼ばれAかZの名前に区別をする添字のついたA(5), B(50)という変数があります。



この変数にはさらに2種類あり、数値を入れる数値変数と文字列を入れる文字変数があります。今迄でできた変数は計算をするための数値を入れておくので、数値変数でしたが、この他にAからZの変数名に\$ (ドルマーク)をつけたA\$, B\$, C\$と書いて使う文字変数があります。PB-400にはこの他に、専用文字変数(\$)という特別な文字変数があります。

数値変数には10桁(仮数部10桁、指数部2桁)までの数値が入り、文字変数には7文字までの文字列が入ります。また、専用文字変数には30文字までの文字が入ります。




この2種類の変数に入れるものが異なりますので、数値変数に"ABC"のような文字を入れることはできませんし、文字変数に数値を入れることもできません。

これ等の変数は用途により使い分け、計算のための数値を入れるのであれば数値変数を使い、メッセージや記号を入れるのであれば文字変数を使います。配列変数は、多くの変数を使いデータを記憶していくときに便利で、1番目、2番目……というように順番で示される番号で変数を区別します。配列変数については、プログラム中で使いながら説明していきますので、ここでは変数の種類として覚えておいてください。


〈変数使用上の注意〉

PB-400では、数値変数と文字変数は変数名が同じであれば同じ箱を使っているのです。



このため、1つのプログラム内でAという数値変数とA\$という文字変数を同時に使えません。

また、配列変数を使うときも同じ箱を色々な名前で呼ぶことがありますので、重ねて使わないように注意してください。



$A=A(0)=A\$=A\(0)

$B=A(1)=B(0)=B\$=A\$(1)=B\$(0)$

$C=A(2)=B(1)=C(0)=C\$=A\$(2)=B\$(1)=C\(0)

⋮

$N=A(13)=B(12)=\dots\dots=N(0)=N\$=A\$(13)=\dots\dots N\(0)

⋮

$Z=A(25)=B(24)=C(23)=\dots\dots=Z\$=A\$(25)=\dots\dots Z\(0)

DE FM文により増設したメモリーについても同じことがいえます。

3-2-3 プログラムの入力

ここでは、組み立てたプログラムを実際にPB-400に覚えさせて(入力して)みましょう。

PB-400に入力するプログラムは、前に出てきた2つの数値を入力して四則計算をさせるプログラムを使います。

プログラム

```
10 INPUT A,B
20 C=A+B
30 D=A-B
40 E=A*B
50 F=A/B
60 PRINT C,D,E,F
70 END
```

このプログラムをPB-400に記憶させるわけですが、まず記憶させる準備をします。

電源スイッチをオンにした状態は **RUNモード** といい、マニュアル計算やプログラムの実行を行なうモードですので、このモードを **WRTモード** (プログラム記憶モード) に切り換えます。プログラムを記憶させることを別のいい方で「書き込む」ともいいますので、WRTモードは「書き込みモード」ともいいます。

では、WRTモードに切り換えるために、**MODE** **1** と押してください。

● **EXE** キーを押すまえにまちがいに気付いた

このときは、**EXE** キーを押さずに、**☞☞** キーを使ってまちがえた箇所にカーソルを合わせ、訂正をします。

例1) `10 INPUT S_` "A"と押すところと"S"と押してしまった。

☞ キーを1回押して"S"にカーソルを合わせる。

☞

正しいキーを押します。

A

続けて正しい操作をして**EXE** キーを押す。

SHIFT **→** **B**

EXE

`10 INPUT S`

`10 INPUT A_`

`10 INPUT A,B`

例2) `PRINT C,,E,F_` 行番号60の"D"を抜かしてしまった。

☞ キーを4回押して、挿入したい箇所の次に合わせます。

☞☞☞☞

1文字分開けます。

SHIFT **INS**

"D"を入れます。

D

訂正が終れば**EXE** キーを押します。

EXE

`60 PRINT C,2`

`60 PRINT C,_`

`0 PRINT C,D2`

`60 PRINT C,D`

例3) `40 EE=A*B_` "E"を1文字多く入れてしまった。

☞ キーを5回押して"E"に合わせます。

☞☞☞☞☞

1文字分削除しますので、**DEL** キーを1回押します。

DEL

削除が終了したら、**EXE** キーを押します。

`40 EE=A*B`

`40 E≡A*B`

`40 E=A*B`

● **EXE**キーを押してからまちがいに気付いた。

EXEキーを押した後ではプログラムとして記憶されてしまいますので、“**LIST**”コマンドを使って再び呼び出し、正しく訂正します。

例) 行番号50を“50 F=A/N”とまちがえてしまった。

LISTコマンドで行番号50を呼び出す。

SHIFT LIST 50 **EXE**

50 F=A/N_

L I S T と押しても同じです

⇐キーを1回押して、“N”に合わせます。

⇐

50 F=A/N

正しいキーを押します。

B

50 F=A/B_

訂正が завершиましたら、必ず **EXE**キーを押します。

EXE

もし行番号60以降が記憶されていれば行番号60が表示されます。

60 PRINT C,D

このほかに訂正がなければ **AC**キーを押して表示をクリアーします。

AC

EXEキーを押しても、このようにLISTコマンドで呼び出して訂正ができます。なお、新たに行番号をつけて書き直すこともできます。すでにある行番号をつけて記憶させた場合は、後から記憶させた方が優先され、前に記憶させた行は消えます。

このようにしてプログラムを記憶させますが、記憶の操作が終了したら、**MODE** **⇐**と押してRUNモードにします。WRTモードのままでは記憶させたプログラムを消したり、書きかえてしまったりすることがありますので、記憶の操作終了後は必ずRUNモードにしてください。

ワンポイントレッスン

プログラムエリア

PB-400にはプログラム分割機能として、10組のプログラムエリアがあります。このプログラムエリアはP0、P1、P2……P9とあり、使い方は同じです。プログラムエリアの指定は**SHIFT**キーに続いて**⇐**から**⇐**のキーを押すことにより行なえます。プログラム指定はRUNモードとWRTモードの両方でも行なえますが、RUNモードで指定した場合はそのプログラムエリアに記憶されているプログラムが自動的にスタートします。WRTモードで指定した場合はプログラムはスタートせずに、プログラムエリアの指定のみ切り換わります。

プログラムエリアの区別は、はっきりとしてください。プログラムを記憶させるときやカセットテープに記録したり、カセットテープから呼び戻したりするときに、別のプログラムエリアで操作しますと正しく行なえません。

電源スイッチをオンにしたときや、ACキーによるオートパワーオフ解除ではP0のエリアが指定されています。

確認の仕方はMODEと押して、“READY”の後に出ている数字により確認できます。

例) MODE → READY P3……プログラムエリア P3

3-2-4 プログラムの実行

では、前に記憶させたプログラムを使って、実際に計算させてみましょう。

MODEと押してRUNモード(“RUN”点灯)であることを確認してください。

プログラムを実行させるには2つの方法があります。

①プログラムエリア指定による実行

SHIFTキーに続いて0～9の数字キーを押しますと、プログラムエリアの指定となり、プログラムが記憶されていれば、プログラムをスタートさせます。

例) SHIFT P0

②RUNコマンドによる実行

プログラム実行開始コマンド“RUN”により実行開始

例) SHIFT RUN (RUNとしても同じ) EXE

この2つの実行方法の違いは、①の方法では必ずプログラムエリアの先頭から実行を開始し、②の方法では先頭からも、任意の行番号からの実行もできることです。

まず、①の方法で実行します。

操 作

SHIFT P0

ここで2つのデータを入力します。

例) 45 EXE

36 EXE

2つのデータを入力すると、和が表示されます。次の答を表示させるにはEXEキーを押します。

EXE

EXE

EXE

表 示

?

?

81

PRINT文を実行すると、
“STOP”が点灯します。

9

1620

1.25

今度はRUNコマンドにより実行してみましょう。

ただし、RUN **EXE**と操作すると①の方法と同じ結果になりますので、20行目から実行してみましょう。

操 作

SHIFT **RUN** 20 **EXE**

(RUN 20 **EXE**でも同じです)

EXE

EXE

EXE

81

9

1620

1.25

このように、RUNコマンドでの実行は、行番号を指定しなければ先頭から実行を開始し、行番号を指定すれば、指定行番号から実行を開始します。

実際にはこの2つの実行方法にはもう一つの違いがあります。

次の操作をしてみてください。

操 作

SHIFT **P5**

P5のエリアにプログラムが記憶されていなければ、何も表示されません。

ここでRUNコマンドにより実行させてみます。

SHIFT **RUN** **EXE**

(RUN **EXE**でも同じです)

RUNコマンドの実行では、現在設定されているプログラムエリア (この例ではP5)のプログラムを実行させます。もしこのP5に設定されているときに、P0のプログラムを実行したい場合はどうすればよいでしょうか。

それは、**SHIFT** **P0**と操作して、プログラムエリアを指定するのです。

SHIFT **P0**

?

このように、2つの実行方法には違いがありますので、用途に合わせて使い分けてください。

プログラムを作り上げ、PB-400に記憶させた後、すぐにでも実行させてみたいものです。もし、実行させてエラー(ERR表示)になっても、がっかりしないでください。こんなときには、次の項目を読んでまちがい探し(デバッグと言う)を行なってください。

ステップ数のかぞえ方

PB-400の持っているプログラム容量は、全部で3616ステップです。

このステップとはプログラムを記憶できる許容量を示す単位で、プログラムを記憶させていく、残りステップ数が減っていきます。

現在の残りステップ数は、**MODE** **1** と押してWRTモードにすると表示されます。

例)

MODE 1 WRT 3616 残りステップ数
P 0123456789

また、使用するステップ数は次のようにかぞえます。

- 行番号……………1~9999のいくつであっても2ステップ
- コマンド……………1ステップ
- 関数……………1ステップ
- 文字……………1文字で1ステップ
- 以上の他に、区切りとして **EXE** キーを押して記憶させるのに1ステップ必要とします。

例) 1 INPUT A 5 ステップ

 $\frac{1}{2} + \frac{1}{1} + \frac{1}{1}$

10 B=SIN A 7ステップ°

$$\begin{array}{ccccccc} \text{---} & \text{---} & \text{---} & \text{---} & \text{---} & \text{---} & \text{---} \\ 2 & 1 & 1 & 1 & 1 & 1 & 1 \end{array}$$

100 PRINT "B=" ; B 10ステップ

2 1 4 1 1 1 計22ステップ

- メモリーを増設した場合は、1つにつき8ステップ必要とします。

例) 標準状態……26メモリー、3616ステップ°

DEFM 10 **EXE**.....36メモリー、3536ステップ°

3-2-5 デバッグ(まちがいを直す)

プログラムを作り上げ、いざ実行開始というときに、実行させてもエラーが表示されたり、結果が思うように得られない。このような事はよくあることです。ので、がっかりせずに、原因を追求してください。

プログラム中にある「まちがい」のことを「バグ」(虫の意味)と呼び、この虫を取り除くという意味で「デバッグ」といいます。


このデバッグの方法も原因により異なります。実行中にエラー表示になる場合と、エラーにはならないが結果が思うように得られない場合です。

実行中にエラーが表示される場合は、PB-400がエラーの発生した箇所とエラーの内容を知らせてくれますので、原因追求は比較的簡単にできますが、エラーは表示しないが結果が思うように得られない場合は、けっこう厄介なものです。

では、順を追ってデバッグをしてみましょう。

(1) エラー表示によるデバッグ

エラー表示とは、エラーメッセージとも呼ばれ、



このように、3つの要素をおしえてくれます。

エラー内容は、「ERR」に続くコードナンバーにより、どのようなエラーが起きたかを知らせてくれます。このコードナンバーは1から9の数字で、「ERR1」は「メモリーオーバー」とか、「ERR2」は「構文エラー」というような内容が決っています。このコードナンバーの内容については、巻末136ページの「エラーメッセージ一覧表」を参照してください。

エラー発生プログラムエリアは、エラーの発生したプログラムエリアを知らせてくれます。

エラー発生行番号は、エラーの発生した行番号を知らせてくれます。

この3つの要素により「どこで」、「どのような」エラーが起きたのかを知ることができます。

それでは例を上げ説明していきましょう。

よく起きるエラーに「ERR2」があります。これは「構文エラー」といい、プログラムを記憶させるときに、まちがって記憶させてしまうと起きるのです。

前の例題で使ったプログラムをまちがえて記憶させてみます。

操 作

MODE 1
SHIFT LIST 20 EXE
← ← DEL
EXE
AC MODE

表 示

P _123456789
20 C=A+B
20 C=AB
30 C=A-B
READY P0

ここでは、行番号20の“C=A+B”を“C=AB”とまちがえました。
では実行させましょう。

操 作

SHIFT P0
45 MODE
12 EXE

?
?
ERR2 P0-20

ここでエラーメッセージが表示されます。これは、プログラムエリアP0の行番号20で、構文エラーが発生しましたという意味です。
では、行番号20を調べてみましょう。

操 作

AC……エラー解除
MODE 1
SHIFT LIST 20 EXE

表 示

—
P _123456789
20 C=AB

ここで正しいプログラムと合っているかをチェックします。
AとBの間の“+”が抜けていますので、正しく直します。

操 作

←
SHIFT INS
+ EXE
AC MODE

表 示

20 C=AB
20 C=A_B
30 C=A-B
READY P0

このように、“ERR2”はプログラムの入力ミスが多いので、“ERR2”が表示されたときはエラーの発生した行番号のプログラムをよくチェックしてください。
なお、記憶のまちがいがだけでなく、READ文(67ページ参照)でデータを読み込むときに、数値変数に文字データを読み込もうとすると“ERR2”が表示されます。“ERR2”の発生した場所にREAD文があるときは、DATA文中のデータもチェックしてください。

では、エラーの種類別にチェックポイントを上げてみましょう。

●ERR1：メモリー不足。スタックオーバー。

残りステップ数を確認して、DEFM文でステップ数以上をメモリーに変換しようとしたかチェックする。

●ERR2：構文エラー

記憶させたプログラムにまちがいがいないかチェックする。

●ERR3：数学的エラー

数式の演算結果が 10^{100} 以上であったり、関数の入力範囲をこえていないかをチェックする。特に変数を使っている場合は、前後関係から変数の内容をチェックする。(0による除算や、負数の平方根をとっている場合が多い。)

●ERR4：未定義行番号エラー

GOTO文やGOSUB文、RESTORE文による行番号の指定が正しくないで、行番号を確認する。

●ERR5：引数エラー

引数やパラメータを必要とするコマンドや関数において、引数やパラメータの値をチェックする。特に変数を使っている場合は、前後関係から変数の内容をチェックする。

●ERR6：変数エラー

配列変数を使うときに、DEFM文でメモリーの増設を行なっているかチェックする。また、同じメモリーを文字変数と数値変数の両方に、同時に使っていないかチェックする。

●ERR7：ネスティングエラー

エラーの起きた行がRETURN文やNEXT文であれば、正しくGOSUB文やFOR文に対応しているかチェックする。また、エラーの起きた行がGOSUB文やFOR文であれば、ネスティングのくり返しをチェックして、GOSUB文は8回まで、FOR文は4回までとする。

●ERR8：パスワードエラー

パスワードが設定されているときに、別のパスワードを入力しようとしたり、使えないLISTコマンドや、NEW、NEW ALL等の操作をしたかチェックする。

●ERR9：オプションエラー

カセットインタフェイス<FA-3>やミニプリンタ<FP-12>が正しく接続されているかチェックする。<FA-12>に充電されているか、または紙づまりをしていないかチェックする。<FA-3>に接続しているテープレコーダーの音量調整やトーンの調整を変えて再び行なったり、テープレコーダーのヘッドを掃除したり、テープを新しいものと変えてみる。録音のときは白プラグのみを、再生のときは黒プラグのみを差し込んで試してみる。

以上が、エラーに対するチェックポイントです。ここに出てきたコマンドについては、後で順を追って説明していきますが、詳しい説明については87ページからの「コマンドリファレンス」編を参照してください。

(2) エラーは表示されないが、結果が思うように得られない。

このようなデバックは、プログラム中の計算式や変数の与え方がまちがっている場合が多いので、計算式や変数の動きをチェックする必要があります。特に計算結果が思うように得られない場合は、元となる公式や計算式と比べてチェックしてください。

プログラムを実行したら止まらなくなったときや、作業をせずに終わってしまうときには、プログラムの流れを制御する変数の動きをチェックしてください。計算式のチェックは、前例(1)のようにWRTモード(MODE 1と押す)で、計算式の書き込んである行をチェックします。

プログラムの流れをチェックする場合は、制御する変数にデータを入れた後にSTOP文で停止させたり、PRINT文により変数の値を表示させてチェックします。

次のプログラムを記憶させてください。

```
10 INPUT A
20 B=1
30 FOR C=1 TO A
40 B=B*C
50 C=C+1
60 NEXT C
70 PRINT B
80 END
```

このプログラムはINPUT文により入力したデータの階乗を求めるプログラムで、本当は行番号50は不要で、このようにFORループ用の変数を変化させてはいけません。

なお、行番号30と60のFOR・NEXT文はくり返し計算を行なわせるためのループ命令です。このFOR・NEXT文については後で説明しますので、ここではとりあえず、記憶させてください。

操 作

MODE 1

SHIFT P1

NEW EXE

└ プログラムを消す命令です。

10 SHIFT INPUT A EXE

20 B=1

表 示

P	1	2	3	4	5	6	7	8	9
P	1	2	3	4	5	6	7	8	9
P	1	2	3	4	5	6	7	8	9


10	INPUT	A
20	B=	1


```

30  [SHIFT] [FOR] C=1 [SHIFT] [TO] A [EXE]
40  B=B*C [EXE]
50  C=C+1 [EXE]
60  [SHIFT] [NEXT] C [EXE]
70  [SHIFT] [PRINT] B [EXE]
80  END [EXE]

```

30	FOR C=1 T
40	B=B*C
50	C=C+1
60	NEXT C
70	PRINT B
80	END

これでプログラムの記憶は完了しましたので、**AC MODE**  と押して RUN モードにします。

では実行してみましょう。

操 作

[SHIFT] P1

例) 12 **EXE**

表 示

?
10395

本当の答は、479001600です。

ここで計算式をチェックしますが、計算式は合っています。次に FOR・NEXT ループの流れを見てみます。

行番号50の次に STOP 文を入れて、プログラムを 1 回ごとに停止させてみます。

操 作

MODE 1

55 STOP **EXE**

AC MODE 

表 示

P 123456789
55 STOP
READY P1

この STOP 文は行番号 50 次に入れますので、50 と 60 の間を選んで行番号をつけます。ここでは行番号 55 としました。

では実行してみましょう。

操 作

[SHIFT] P1

12 **EXE**

ここでループの制御変数 C の値を見ます。

C EXE



実行を続けます。

EXE

再び変数 C の値を見ます。

C EXE

表 示

?	
	STOP
2	STOP
	STOP
4	STOP

この変数 C の値は 1 から 1 つずつ増えていかなければいけないのに、2 つずつ増えています。これで、FOR・NEXT ループの動き (流れ) が正しくないことがわかります。

プログラムをもう一度見なおして、変数Cの流れを見ます。ここでは行番号50に問題があり、必要ないことがわかりますので、行番号50と後から追加した行番号55のSTOP文を削除します。

操 作

MODE 1

50 EXE

55 EXE

AC MODE 2

表 示

P	1	2	3	4	5	6	7	8	9
-									
-									
READY P1									

これでデバッグは完了です。

このSTOP文によるデバッグの他に、トレースモード (MODE 2) と押す。“TR”点灯) によるデバッグがあります。

トレースモードでのデバッグは、プログラムを1命令ごとに実行して停止します。この停止している間に変数の値を見たりできますので、EXEキーで1命令ごとに進めてデバッグをします。

前例でためしてください。EXEキーを押すごとにプログラムエリアと行番号を表示します。

なお、トレースモードを解除するには、MODE 3 と押し、“TR”を消します。

以上のようにしてデバッグを行ないますので、エラーが表示されたり、思うように結果が得られなくてもがっかりせずに、確実にデバッグを行なってください。


3-3 発展するプログラム

今迄の説明でプログラムについて、だいたいわかってきたと思います。プログラムの基本として、入力、計算、出力の3つのポイントがあります。この3つを使うだけでも色々なプログラムが作れますが、ここで説明するコマンドは、憶えておくともっと便利になったり、使いやすくなったりします。ここでは、使って便利なコマンドについて説明していきますので、順に少しずつ憶えていってください。

3-3-1 プログラムの流れを変える〈GOTO文〉

プログラムの3つの基本に加え、同じ計算を何回も繰り返したり、プログラムの流れを行番号の順ではなく、任意の行番号へ移すときに便利なGOTO文があります。

まず例題として、ある数値の2乗を求めるプログラムを作ってみましょう。このプログラムは「データの入力」、「2乗の計算」、「答の表示」の3つでできますので、フローチャートを作ってみましょう。



このフローチャートにそってプログラムにします。

```
10 INPUT A
20 B=A*A
30 PRINT B
40 END
```

プログラムの記憶方法は、もうわかったと思いますので、ここからは省略します。もしわからなくなったときは、32ページの「プログラムの入力」編をご覧ください。

ここでは例として、15と43の2乗を求めてみましょう。

操 作

RUN **EXE**

15 **EXE**

RUN **EXE**

45 **EXE**

表 示

?
225
?
1849

このように、1回ごとに実行(RUN **EXE**)させないとできません。データがいくつもあるような場合には、とても不便です。

この計算が何回も繰り返しできれば便利だと思いませんか？

この繰り返しを便利にする命令がGOTO文です。

GOTO文の働きは、GOTOの後に示される数値に該当する行番号やプログラムエリアに、プログラムの流れを移します。

ここで使っているプログラムにGOTO文を加えてみましょう。それは、行番号40のEND文をGOTO文にします。

40 GOTO 10

この意味は、以後の流れを行番号10に移します。(ジャンプします)

さっそく、記憶させたプログラムを変更して実行してみましょう。

操 作

RUN **EXE**

15 **EXE**

EXE

43 **EXE**

表 示

?
225
?
1849

このように、GOTO文は繰り返し計算に便利な命令です。

GOTO文はプログラムの先頭に戻して、繰り返し実行させるだけでなく、任意の位置にジャンプすることができますので、他にも便利に使えます。

たとえば、次のプログラムを見てください。

10 INPUT A

20 GOTO 50

30 PRINT B


40 GOTO 10

50 B=A*A

60 GOTO 30

このプログラムはとてもおかしいプログラムですので、実行させなくてもかまいませんが、使い方は前のプログラムと同じです。

このプログラムの流れは次のようになっています。



このように、GOTO文は指定された行番号に無条件にジャンプしますので、別名“無条件ジャンプ”とも呼ばれています。

GOTO文では、行番号へのジャンプの他に、プログラムエリアへのジャンプもできます。GOTOの後に“#”を付けて0～9の1文字でプログラムエリアを指定します。

例) GOTO #1……プログラムエリアP1にジャンプする。

GOTO #9……プログラムエリアP9にジャンプする。

プログラムエリアへのジャンプでは、そのエリア内のプログラムの先頭から実行を続けます。

ワンポイントレッスン

PRINT文

PRINT文は、後に続く変数の内容や文字列、数値を表示する命令です。

変数は数値変数でも文字変数でもよく、変数の内容を表示します。

例) A=123のとき……PRINT A→123

B\$="ABC"のとき……PRINT B\$→ABC

文字列を"で囲んで書けばそのまま表示されますので、メッセージとして使うこともできます。

例) PRINT "CASIO"→CASIO

表示させたいものが2つ以上ある場合は、,または;で区切ることにより、続けて書けます。

例) PRINT A,B,Z\$

PRINT "TOTAL=";T

区切りの,と;の違いは、,では1つ目の内容を表示後“STOP”を点灯して止まり、**EXE**キーを押すことにより次の内容を表示します。区切りが;のときは1つ目を表示後、続けて一行の中に表示します。

例) 次のプログラムで試してください。

```

10 A=123
20 B$="ABC"
30 PRINT A,B$.....Aの内容を表示後、EXEキーを押すことにより
                        B$の内容を表示。
40 PRINT A;B$.....AとB$の内容を続けて表示。
50 PRINT B$; .....B$の内容を表示後、停止せずに次に進む。
60 PRINT A .....Aの内容を表示後、停止する。
70 END

```

このプログラムを実行しますと、次のようになります。

操 作	表 示
RUN EXE	123
EXE	ABC
EXE	123ABC
EXE	ABC 123


} PRINT A,B\$
 ...PRINT A;B\$
 ...{ PRINT B\$;
 ...{ PRINT A

区切りが;のときに続けて表示させても、数値の前に1文字分あいて見えます。これは符号(+,-)を表示しているのですが、正の+符号は省略されて空白(スペース)として表わされるので、あいているように見えるのです。

〔練習問題〕

任意の半径を入力して、円の面積を求め、GOTO文により繰り返すプログラムを作る。 公式： $S = \pi r^2$ (π は $\text{SHIFT} + \pi$ と押す)

まず、フローチャートは次のようになります。変数は公式にそってS、Rを使います。



プログラム

10 INPUT R	10 INPUT R	
20 S=π*R*R	20 S=π*R↑2	2乗計算は*↑2"とも書きます。
30 PRINT S	30 PRINT S	
40 GOTO 10	40 GOTO 10	

または

3-3-2 プログラムに判断させる<IF~THEN文>

プログラム中で大小の判断や自動的に制御ができれば、とても便利なものです。
この判断をプログラム中で行なう命令がIF文です。

IF文は"IF"と"THEN"の間の比較式により判断します。

IF 比較式 THEN { 行番号またはプログラムエリア
 命令文 }


比較式の比較結果により、正しければ"THEN"以降の行番号またはプログラムエリアにジャンプしたり、命令文を実行します。正しくなければ、次の行から実行を続けます。

では実際に、IF文の働きを見てみましょう。

例) 任意の数を入力して、10より大きければ再び入力に戻り、10より小さければその値の2乗を求め、答を表示して再び入力に戻る。


このプログラムを分析しますと、「入力部」、「判断部」、「計算部」、「表示部」の4つからできています。

判断部のフローチャート記号は次のものを使います。



この記号の"YES"と"NO"の位置は逆でもかまいませんが、わかりやすくするために、"YES"、"NO"を付けておきます。

では、フローチャートにしてみましょう。



フローチャートの左側に付いている数字は行番号で、このようにしておくと、すぐにプログラムにすることができます。

```
10 INPUT A
20 IF A>10 THEN 10
30 B=A*A
40 PRINT B
50 GOTO 10
```

行番号20がIF文の使い方で、比較式の結果が正しければTHEN以降を実行しますので、この場合はTHEN 10で、行番号10へジャンプします。

比較式に用いる比較記号には次のものがあります。

左辺>右辺……右辺より左辺が大きい
左辺<右辺……右辺より左辺が小さい
左辺=右辺……右辺と左辺が等しい
左辺≥右辺……右辺より左辺が大きいか等しい
左辺≤右辺……右辺より左辺が小さいか等しい
左辺≠右辺……右辺と左辺が等しくない

また、“THEN”は“GOTO”の意味も含まれていますので、“THEN GOTO 10”は“THEN 10”と書くことができます。

では、このプログラムを実行してみましょう。

操 作

RUN **EXE**

5 **EXE**

EXE

12 **EXE**

9 **EXE**

表 示


?
25
?
?
81

このように、IF文により判断して、データを選ぶことができます。

例) いくつかのデータを入力して、“0”を入力したときにデータの平均を求める。

このプログラムを分析しますと、「入力部」「判断部」「計算部」「表示部」に分けられます。しかし計算部は平均を求めますので、「合計を求める」「個数を数える」「平均を求める」の3つがあります。この3つのうち、平均を求める計算だけは“0”を入力したときだけ実行すれば良いので、判断の後に続くと考えられます。

この分析にもとづいて、フローチャートを作ってみます。



このフローチャートからわかるように、データを変数Aに入れ、変数Aが0であるかをIF文により判断し、0でなければ合計とデータ数を求めて、再びデータ入力に戻ります。もしAが0であれば平均を表示させ、終了します。ただし、1回目の入力が0であると0で割り算をすることになるので、エラーとなりますから、注意してください。

この例は前回より少しむずかしくなっていますので、ゆっくりと考えてください。まず計算部ですが、合計計算は合計を入れておく変数を決め、その変数に入力したデータを加えていく方法がとられます。ここでは合計用の変数をBとしていますので、“ $B = B + A$ ”という計算になります。(変数Bの内容に変数Aの内容を加えて、変数Bとします) データ数は1個のデータにつき1つつカウントしていけばよいので、カウント用の変数(ここではC)に1つつ加えていけばよいのですから、“ $C = C + 1$ ”となります。

では、肝心な判断部を見ますと、Aが0であれば平均を表示しますので、PRINT文により平均の計算結果を表示させます。PRINT文では変数の値の他に計算式を書けば、計算式の結果(答)を表示しますので、“PRINT B/C”となります。IF文ではTHENの後に命令文も書けますので、THENの後に“PRINT B/C”を続けて書きます。

ここで1つ問題になることがあります。それは変数Bと変数Cで、このままプログラムにすると、前に何かの値が入っていればどんどん加算されていきますので、答が違ってしまいます。そこで変数Bと変数Cに0を入れておかねばなりません。これは“ $B = 0$ ”“ $C = 0$ ”となります。この2つの代入式に別々の行番号をつけてもかまいませんが、短いプログラムはマルチステートメントという“:”で続けて、“ $B = 0 : C = 0$ ”と書くことができます。

では、プログラムにしてみましょう。

```

10 B=0:C=0
20 INPUT A
30 IF A=0 THEN PRINT B/C
40 B=B+A
50 C=C+1
60 GOTO 20

```

これでプログラムができましたが、平均を表示させてもプログラムは終了しません。そこで、行番号30のPRINT文の後にマルチステートメントでEND文を加えます。

```

30 IF A=0 THEN PRINT B/C:END

```

このように、IF文は比較式により判断を行ない、その結果によりプログラムの進行を決めます。

●IF文の応用

今迄出てきた例では、1つの判断によりプログラムの進行を決めていましたが、もし判断が2つ以上あり、全ての条件を満たさなければならないときはどうしましょう。

たとえば、任意の数値を入力し、1～9のものだけを選ぶとします。

これを別の考え方をすれば、0より大きく、10より小さいということですので、“0<変数”と“変数<10”の2つの比較が必要です。

これを1行に書くとすれば次のようになります。

```

IF 0<変数 THEN IF 変数<10 THEN.....

```

比較条件が3つ以上の場合も同様にできますが、複雑になりすぎたり、1行が長すぎたりしますので、2つ位にしておくのが良いと思います。

ワンポイントレッスン

マルチステートメント(:)

マルチステートメントは短い代入式などを1行にまとめたり、IF文のTHEN以降でいくつもの命令があるときなどに便利です。

```

例1) 10 A=1
      20 B=2 } ⇒ 10 A=1:B=2:C=3
      30 C=3 }
           ⋮
           ⋮
           ⋮

```

例2)
$$\left. \begin{array}{l} 50 \text{ } C=A+B \\ 60 \text{ } D=A-B \\ 70 \text{ } E=A*B \\ 80 \text{ } F=A/B \\ \vdots \end{array} \right\} \Rightarrow 50 \text{ } C=A+B:D=A-B:E=A*B:F=A/B$$

IF 文中で THEN 以降にマルチステートメントを使う場合は、比較式が正しいときにだけマルチステートメント以降を実行しますので、注意してください。

例3) IF A<0 THEN A=10:B=20:.....

比較が正しいときにだけ実行します

マルチステートメントはプログラムをまとめるのに便利ですが、あまり1行を長くしすぎると逆に見づらくなりますので、適当な長さで次の行に分けて書いてください。


〔練習問題〕

入力した数値を 0 より大きい数と小さい数に分けて合計を求めるプログラムを作る。

<ヒント>

数値入力後、IF 文により 2 つの変数のどちらかに合計させる。

なお、合計を取る変数はあらかじめ 0 を代入しておく。



プログラム

```
10 A=0:B=0
20 INPUT C
30 IF C>0 THEN A=A+C:GOTO 50
40 B=B+C
50 PRINT A;B
60 GOTO 20
```

変数AとBに0を代入しておきますが、マルチステートメントで続けなくてもかまいません。

行番号30のIF文では、入力した値(変数Cの値)が0より大きいかを判断して、0より大きいときはTHEN以降で変数Aに合計をとり、そうでないときは行番号40で変数Bに合計をとります。

行番号50では入力することに各々の合計を表示させます。

3-3-3 プログラムを繰り返す<FOR・NEXT文>

ある一定の回数だけ計算などを繰り返す場合に、GOTO文やIF文の組み合わせではプログラムが長く複雑になってしまいます。繰り返しの回数がわかっているときなどは、もっと簡単にプログラムを組みたいものです。そこで、この繰り返しを簡単に行なう命令がFOR・NEXT文です。

FOR・NEXT文は、FOR文とNEXT文の間にある計算などの作業を指定回数分だけ繰り返します。

FOR文は、

FOR 制御変数=初期値 TO 終値 STEP 刻み幅

という形式です。

NEXT文は、

NEXT 制御変数

という形式です。

FOR文中で、制御変数として使えるのはAやBのような1文字の数値変数だけで、配列変数は使えません。初期値、終値、刻み幅は数式や数値変数で与えることができ、初期値から終値を越えるまで、刻み幅ずつ制御変数を変化させて、繰り返します。刻み幅は省略でき、省略した場合は1となります。

次のプログラムを記憶させて実行すると、FOR・NEXT文の動きがよくわかります。

```
10 INPUT A
20 FOR B=1 TO 10 STEP A
30 PRINT B
40 NEXT B
50 GOTO 10
```


このプログラムを実行しますと、次のようになります。

操 作

RUN EXE

3 EXE

EXE

EXE

EXE

EXE

0.8 EXE

EXE

EXE

EXE

⋮

表 示

?
1
4
7
10
?
1
1.8
2.6
3.4


⋮

このように、FOR・NEXT文は初期値から終値までを刻み幅で繰り返します。なお、制御変数に使う変数名は、ここではBを使っていますが、一般にはI、J、Kがよく使われます。

例) データ数を入力し、データの合計と平均を求めるプログラムを作る。

このプログラムでは、まずデータ数を入力し、その後にFOR・NEXT文により個々のデータを入力して合計を求めます。データの入力が終われば、後は合計と平均を表示させます。

最初にフローチャートを作ってみましょう。



※この記号はFOR・NEXTループを示す記号で、変数Bが1から順に1ずつ増えていき、データ数より大きくなったらFOR・NEXTループから抜け出て、次の作業に移すことを表わします。

このフローチャートをもとにしてプログラムを作ります。

```

10 D=0
20 INPUT A
30 FOR B=1 TO A
40 INPUT C
50 D=D+C
60 NEXT B
70 PRINT D,D/A
80 END

```

このように、データ数がわかっている場合には、FOR・NEXT文により繰り返してデータ入力や計算を行なうことができます。

行番号20のように、データ数をINPUT文により入力せずに、直接、行番号30のAのかわりに書き込むこともできます。このときは行番号20は不要となります。


〔練習問題〕

階乗を計算するプログラムを作る。

〈ヒント〉

$5! = 1 \times 2 \times 3 \times 4 \times 5$ ですので、FOR・NEXT文により回数分ループさせて計算する。

フロチャート



プログラム

```
10 INPUT A
20 C=1
30 FOR B=1 TO A
40 C=C*B
50 NEXT B
60 PRINT C
70 END
```

行番号20は階乗を求める変数Cに1を代入しておきます。これは、階乗計算のときにCの値が初期値でないでと答がかわってくるからです。

行番号30から50のFOR・NEXT文で階乗計算をします。これは、変数Bの値を1から1つつ増やしてゆき、 $1 \times 2 \times 3 \times \dots$ と計算して階乗を求めます。

行番号70のEND文により、1回の計算でプログラムは終了しますが、繰り返し計算するときは、行番号70を“GOTO 10”としてください。


3-3-4 複雑なプログラムに便利なサブルーチン

〈GOSUB・RETURN文〉


プログラム作りに慣れてくると、プログラムが長く複雑になったり、同じ操作をする部分が何回も出てきたりします。このようなときに、プログラムをブロックごとに作り上げ、構造化してスッキリさせたり、同じ部分を共通して使えるようにするのがサブルーチンと呼ばれるものです。

サブルーチンは副プログラムとも呼ばれ、プログラム全体の流れを制御するメインルーチンに対し、部分的作業を受け持つのがサブルーチンです。

メインルーチンのみ



サブルーチンつき




では、実際のプログラム例で働きを見てみましょう。

例) 順列と組み合わせを求めるプログラムを組む


公式： 順 列 ${}_nP_r = \frac{n!}{(n-r)!}$

組み合わせ ${}_nC_r = \frac{n!}{r!(n-r)!}$

このプログラムでは2つのデータ(n 、 r)を入力し、順列と組み合わせを求めます。
まず、簡単フローチャートを作ってみましょう。



このフローチャートの中で、順列と組み合わせの計算の部分をもう少し詳しくしてみましょう。



2つの計算の中で、 $n!$ 、 $(n-r)!$ を求める計算が共通していることに気づくと思います。そして、さらに詳しく見ると、階乗計算が3回行なわれます。この階乗計算は前節の練習問題で行なっていますので、わからない方は前に戻ってもらう一度やりなおしてください。3回の階乗計算をその都度行ないますと、かなり長いプログラムとなります。


```

プログラム例 (1)
10 INPUT N,R
20 A=1
30 FOR B=1 TO N
40 A=A*B
50 NEXT B
60 E=A
70 A=1
80 FOR B=1 TO N-R
90 A=A*B
100 F=A
120 P=E/F
130 A=1
140 FOR B=1 TO R
150 A=A*B
160 NEXT B
170 G=A
180 C=E/G/F
190 PRINT P,C
200 END

```

メモリー内容

N : n
R : r
P : 順列
C : 組み合わせ
A : 階乗用変数
B : FOR・NEXT
ループ用変数
E : $n!$
F : $(n-r)!$
G : $r!$

$n!$ 計算

$(n-r)$ 計算

$\frac{n!}{(n-r)!}$ 計算

$r!$ 計算

$\frac{n!}{r!(n-r)!}$ 計算

このプログラムの中で、3つの階乗計算は共通のプログラムを使っています。異なる点はFOR 文中の終値です。この終値も共通の変数で制御できれば、この3つの計算を完全に共通化できます。この共通化して使う方法がサブルーチンとしての使い方です。

なお、終値に変数Hを使って共通化するには、変数Hに n 、 $n-r$ 、 r の値を前もって入れておく必要があります。

では、このプログラムをサブルーチン方式に変更しますが、サブルーチンに作業を引き渡す命令と、作業終了後に再び戻ってくる命令が必要です。この2つの命令が"GOSUB"と"RETURN"です。

```

プログラム例 (2)
10 INPUT N,R
20 H=N
30 GOSUB 150
40 E=A
50 H=N-R
60 GOSUB 150
70 F=A
80 P=E/F
90 H=R
100 GOSUB 150

```

```

110 G=A
120 C=E/G/F
130 PRINT P,C
140 END
150 A=1
160 FOR B=1 TO H
170 A=A*B
180 NEXT B
190 RETURN

```

} サブルーチン

サブルーチンはこのように共通な部分を持つプログラムを、短かく、スッキリと仕上げてくれます。

この例では1行分しか短くなっていませんが、もっと複雑になってきたり、大きなプログラムを作り上げるうえでは、重要な役割割りを持っています。初めのうちはあまり使わないかもしれませんが、覚えておくと便利に使える命令です。

〔練習問題〕


標準偏差を求めるプログラムを作る。ただし、データ入力部と総和、2乗和、データ数の算出部分はサブルーチンとして作る。

公式：
$$\sigma n = \sqrt{\frac{\sum x^2 - (\sum x)^2 / n}{n}}$$

$\sum x$: 総和
 $\sum x^2$: 2乗和
 n : データ数

〈ヒント〉

データ入力後、IF文により"0"であればメインルーチンに戻り、標準偏差を求める。平方根はSQRを使う。



プログラム

```

10 B=0:C=0:D=0
20 GOSUB 100          .....サブルーチンへ
30 E=SQR((C-B*B/D)/D).....標準偏差
40 PRINT E
50 END
100 INPUT A
110 IF A=0 THEN RETURN
120 B=B+A              .....総和
130 C=C+A*A            .....2乗和
140 D=D+1              .....データ数
150 GOTO 100

```

} サブルーチン

このプログラムでは、行番号20でサブルーチンへ作業を引き渡し、行番号100からのサブルーチンでデータ入力と総和、2乗和、データ数のカウントを求めています。


行番号110のIF文はデータ入力終了の判断で、0を入力すると“THEN”以降に進み、メインルーチンへ戻ります。

なお、行番号50のように、メインルーチンの最後には必ずEND文を入れてください。

3-3-5 関数を使う

マニュアル計算でも使いましたが、プログラム中に関数を書き込んでも使えます。プログラム中でも使い方は同じですので、ここでは実際のプログラム例としていくつかを説明します。

例1) 三角関数



左の三角形で辺Aと角 α を入力して、辺Bの長さを求める。

公式： $B = A \cdot \sin \alpha$

単位：度

プログラム例

```

10 INPUT A,D
20 MODE 4
30 B=A*SIN D
40 PRINT B
50 END

```

操作例

RUN EXE
 (辺A) 15 EXE
 (角 α) 30 EXE


表示

?
?
7.5

行番号20は角度単位の指定で、ここでは度で計算しますので“MODE 4”となります。

行番号30で三角関数を使って計算します。

例2) 三角関数



左の円で、半径 r と角度 θ を入力して点 P の座標 (x, y) を求める。

$$\text{公式: } x = r \cdot \cos \theta$$

$$y = r \cdot \sin \theta$$

単位: ラジアン

プログラム例

```
10 INPUT R, T
20 MODE 5
30 X=R*COS T
40 Y=R*SIN T
50 PRINT X, Y
60 END
```

操作例

RUN **EXE**
(半径) RU5 **EXE**
(角 θ) $\pi / 3$ **EXE**

表示

?
?
17.5

角度単位がラジアンですので、行番号20で MODE 5 を指定します。

行番号30と40で x 座標、 y 座標を求めます。

例3) 逆三角関数



左の三角形より、辺 A 、 B を入力して角度 α を求める。

$$\text{公式: } \alpha = \tan^{-1} \frac{B}{A}$$

単位: 度

プログラム例

```
10 INPUT A, B
20 MODE 4
30 D:ATN(B/A)
40 PRINT D
50 END
```

操作例

RUN **EXE**
(辺 A) 100 **EXE**
20 **EXE**

表示

?
?
14.5742162

例4) 10進↔60進変換

時間計算をするプログラムを作る

プログラム例

```
10 T=0
20 INPUT D, E, G
30 S=SGN D
40 D=ABS D
50 T=T+S*DEG(D, E, G)
60 PRINT DMS$(T)
70 GOTO 20
```

操作例

RUN **EXE**
(時) 1 **EXE**
(分) 25 **EXE**
(秒) 36 **EXE**
EXE
(時) 2 **EXE**
(分) 15 **EXE**
(秒) 5 **EXE**

表示

?
?
?
1° 25' 36
?
?
?
3° 40' 41

行番号20では時・分・秒を3つの変数D、E、Gに入力します。

行番号30と40は減算するときのために、時に負符号(－)をつけて入力すれば、前回までの結果から次の時間を減算します。もし、加算だけを行なう場合は不要です。

行番号50で合計を求めますが、変数Sには加算のときは1が、減算のときは－1が入っていますので、加算・減算ができます。DEG関数は60進数の時・分・秒を10進数に変換する関数で、合計計算は10進数で行なわれます。

行番号60は表示用で、10進数を60進数に変換するDMS\$関数を使って表示します。

例5) 乱数発生

1 から99までの乱数を作る。

プログラム例

```
10 R=INT(RAN#*99)+1
20 PRINT R
30 GOTO 10
```

行番号10で乱数を作ります。この例では1 から99ですので、RAN#に99をかけて、1を加えます。

乱数が5 から9のとき→INT(RAN#*5)+5

乱数が10から20のとき→INT(RAN#*11)+10

この他にもいくつかの関数がありますが、同じように使えます。

3-3-6 配列を使う

配列とは配列変数のことで、一般のAからZまでの変数の使い方とは異なり、同じ変数名でも番号で管理する変数です。

変数名はアルファベット1文字で、AからZまでが使える、その変数名に管理番号がつきます。

例) A(1)

└───┬───┐
 | |
 | └─── 管理番号(添字)
 └─── 変数名

配列変数は、多量のデータを扱うときに便利ですので、使い方を充分覚えてください。

たとえば、10個のデータを入力するには、

```
FOR A=1 TO 10
INPUT N(A)
NEXT A
```

50個のデータの中から一番大きいものを選び出すには、

```
A=0
FOR B=1 TO 50
  IF P(B)>A THEN A=P(B)
NEXT B
PRINT A
```

配列変数を使用するときは、配列の取り方に注意してください。配列変数として使う変数と、一般の変数として使う変数の箱が、一部共通に使う所がありたとうば変数名をAとした場合、A(0)はAと同じ箱で、A(1)はBと同じ箱……というように共通の箱を使っていますので、A(0)とAを同じプログラム内で使うと変数内のデータがかわってしまいます。

共通している部分は次のようになっています。

A	B	C	D	X	Y	Z
A(0)	A(1)	A(2)	A(3).....	A(23)	A(24)	A(25)
	B(0)	B(1)	B(2).....	B(22)	B(23)	B(25)
	⋮				⋮	
				X(0)	X(1)	X(2)
					Y(0)	Y(1)
						Z(0)

このために、配列変数により多量のデータを扱う場合は、その他の必要な変数と重ならないように注意してください。

配列変数を使用するときのもう一つの注意点は、**メモリーの増設**です。一般変数で使うAからZと共通する部分を使う場合は、増設する必要ありませんが、それ以上の変数を使う増設は必ず増設を行なってください。

メモリーの増設は**DEFM命令**により行ないます。

DEFM命令は、1個単位でいくつ増設するかを指定します。

たとえば、10個を増設する場合、

マニュアルでは、DEFM 10 **EXE**

プログラム中では、行番号DEFM 10

この命令はマニュアルでもプログラム中に書き込んでも使えますが、配列を使うときなどは、プログラムの最初の方にこの命令を書き込んでおいてください。では、実際に配列変数を使ったプログラムを見てみましょう。

例) 配列変数 $G(0) \sim G(99)$ に 0 から 99 の乱数を入れ、大きい順に並びかえて表示する。

プログラム例

```
10 DEFM 80
20 FOR B=0 TO 99
30 G(B)=INT(RAN#*100)
40 NEXT B
50 BEEP 1
60 FOR B=0 TO 99
70 A=-1
80 FOR C=B TO 99
90 IF G(C)>A THEN A=G(C):D=C
100 NEXT C
110 E=G(B):G(B)=G(D):G(D)=E
120 NEXT B
130 BEEP 0
140 FOR B=0 TO 99
150 PRINT G(B)
160 NEXT B
170 END
```

乱数により
 $G(0) \sim G(99)$ に
0 ~ 99 を代入する


大きい順に並びかえる

順番に表示する。

このプログラムは、大きくわけて 3 つの部分からできています。1 つ目は乱数により 0 から 99 の数値を作り出し、配列変数 $G(0)$ から $G(99)$ に代入します。2 つ目が並びかえで、大きい順に並びかえます。3 つ目は表示で、並びかえ終ったデータを大きい順に表示します。

行番号 50 と 130 の BEEP 文は、ブザー音を発生させるためで、“BEEP 1” で高音を、“BEEP 0” で低音を発生します。この 2 つの行は、データの作成と並びかえ終了のサインとして音を発生させていますので、なくてもかまいません。


行番号 60 ~ 120 が並びかえで、全部のデータの中から 1 番大きいものを選び出して 1 番目の変数に入れ、次に 2 番目以降から最も大きいものを選び出し……という操作を繰り返します。



なお、行番号10のDEFM文はメモリーの増設で、 $G(0) \sim G(99)$ までの100個の変数の内、20個が $G \sim Z$ の変数となりますので、残りの80個を増設します。このように配列変数を用いると、多量のデータを扱うのに便利になります。

〔練習問題〕

10個のデータを入力し、合計と各々のデータの構成比を求める。構成比は百分率で、小数以下2桁まで求める。



プログラム

```
10 A=0
20 FOR B=0 TO 9
30 INPUT G(B)
40 A=A+G(B)
50 NEXT B
60 PRINT A
70 SET F2
80 FOR B=0 TO 9
90 PRINT G(B)/A*100
100 NEXT B
110 SET N
120 END
```

行番号20～50までがデータ入力部で、FOR・NEXT文によりG(0)～G(9)に10個のデータを入力します。行番号70の“SET F2”は、構成比を百分率で小数以下2桁まで表示するための小数以下指定です。行番号80～100で構成比を百分率で表示します。行番号110では、小数以下指定を解除しておきます。

3-3-7 データを読み込む<READ・DATA・RESTORE文>

ここでは、データ処理の一つの方法として、プログラム中にデータを書き込む“READ”、“DATA”、“RESTOR”について説明します。

データの入力方法にINPUT文がありますが、INPUT文はプログラム実行中に“?”を表示してキーボードからデータを入力します。READ文はプログラム中のDATA文に書かれているデータを変数に読み込みます。

例) DATA文にある10個のデータを読み込み、表示する。

プログラム

```
10 FOR B=1 TO 10
20 READ A
30 PRINT A
40 NEXT B
50 DATA 1,2,3,4,5,
6,7,8,9,10
60 END
```

操 作

RUN **EXE**

EXE

EXE

⋮

⋮

⋮

EXE

1
2
3
⋮
10

READ文は、必ずDATA文と対で使われ、“READ”に続く変数に、DATA文から持ってきたデータを代入します。

READ文の後の変数は、(カンマ)で区切ることにより、いくつでもかくことができます。

例2) プログラム

```
10 READ A$,B$
20 PRINT A$,B$
30 END
40 DATA CASIO,PB-400
```

操 作

RUN 

表 示

CASIO PB-400

DATA文の位置はプログラム中のどこにあってもかまいません。プログラム実行後は行番号の若いDATA文の最初のデータから順に変数に代入していきます。READ文中の変数は、DATA文に書いてあるデータが数値のときは数値変数を、文字のときは文字変数を使ってください。

例) プログラム

```
10 RESTORE
20 READ A$
30 PRINT A$
40 READ B
50 PRINT B
60 RESTORE 100
70 READ C$
80 PRINT C$
90 DATA ABC
100 DATA 123456
110 END
```

操 作

RUN 





表 示

ABC
123456
ABC


RESTORE文で、行番号を指定しないとき、次のREAD文で読み込むデータは、そのプログラム中に書かれている最初のDATA文の最初のデータです。行番号の指定があるとき、次のREAD文で読み込むデータは、指定した行番号のDATA文の最初のデータです。

【練習問題】

札幌、東京、名古屋、大阪、福岡の地名をプログラム中で読み込み、付随するデータを入力して、大きい順に並びかえる。

ただし、地名はG\$(0)~G\$(4)に読み込み、データはL(0)~L(4)に入力する。

フローチャート例



プログラム例

```
10 FOR A=0 TO 4
20 READ G$(A)
30 PRINT G$(A);
40 INPUT L(A)
50 NEXT A
60 FOR A=0 TO 4
70 C=0
80 FOR B=A TO 4
90 IF L(B)>C THEN C=L(B):D=B
100 NEXT B
110 E=L(A):L(A)=L(D):L(D)=E
120 F$=G$(A):G$(A)=G$(D):G$(D)=F$
130 PRINT G$(A);L(A)
140 NEXT
150 DATA SAPPORO,TOKYO,OSAKA,NAGOYA,FUKUOKA
```

メモリー

A : FOR・NEXT 文用
B : FOR・NEXT 文用
C : 最大値
D : 最大値の番号
E : 並びかえ用
F\$: 並びかえ用
G\$(0)~G\$(4) : 地名
L(0)~L(4) : データ

このプログラムは、大きくわけて2つにわけられます。1つ目は行番号10~50までの入力部で、2つ目は行番号60~140までの並びかえ部です。入力部はFOR・NEXT文で5回ループさせる間に、DATA文より地名を読み込み、同時にデータも入力します。行番号30のPRINT文は、次のINPUT文によりデータ入力をする前に、地名をメッセージとして表示させるためです。並びかえは前回に行なったものと同様ですが、ここでは地名とデータの2つを同時に並びかえますので、行番号110と120のようになります。このプログラムでは手間をばくために、並びかえと同時に結果を表示させています。行番号150のデータ文は、どこの位置に入れてもかまいませんが、一般には最後の方に入れておく方が見やすいと思います。


3-3-8 間接指定<ON~GOTO、ON~GOSUB文>

GOTO文とGOSUB文の機能については、もうおわかりだと思います。GOTO文やGOSUB文はプログラム中に直接行番号やプログラムエリア番号を書き込んで指定しますが、処理の都合上、演算結果やデータにより分岐先が異なる場合があります。このような場合にIF文で条件を判断しては大変です。もっと便利な命令はないでしょうか？

分岐先をプログラム中で判断して指定する命令が、間接指定の“ON~GOTO”と“ON~GOSUB”です。

ON~GOTO文とON~GOSUB文は似たような機能を持っています。

例)



“ON”に続く数値変数や計算式の結果により、1であれば1つ目の分岐先に、2であれば2つ目の分岐先に、……と分岐します。変数や計算式の値よりも分岐先の数が少ないときや該当する分岐先がないときは、次の行またはマルチステートメント以降の命令に進みます。

例)

```


10 INPUT A
20 ON A GOTO 100,200,300,400,500
30 PRINT " NO"
40 END
100 PRINT "LINE 100" :END
200 PRINT "LINE 200" :END
300 PRINT "LINE 300" :END
400 PRINT "LINE 400" :END
500 PRINT "LINE 500" :END

```

[練習問題]

角度と4～5の数値を入力し、間接指定で角度単位を指定するサブルーチンへ分岐して、サイン(SIN)の答を求める。

フロチャート



プログラム

```

10 INPUT "KAKUDO",A
20 INPUT "TAN I",B
30 ON B-3 GOSUB 100,200,300
40 PRINT SIN A
50 GOTO 10
100 MODE 4
110 RETURN
300 MODE 6
310 RETURN
300 MODE 6
310 RETURN

```

このプログラムは2つのデータを入力しますので、INPUT文にメッセージを加えて入力しやすくしました。行番号10では角度を変数Aに入力し、行番号20では単位を変数Bに入力します。行番号30ではON～GOSUB文を使い、4～6の数値を1～3に直して分岐先を指定します。

サブルーチンでは各々角度単位を指定して、元に戻ります。

3-3-9 文字を扱う文字関数<LEN、MID\$、VAL、STR\$>

SINやCOSなどは数値を扱う関数ということで、数値関数と呼ばれていますが、この他にも文字を扱う文字関数があります。PB-400には文字関数として"LEN" "MID\$"、"VAL"、"STR\$"があります。

●LEN

LEN関数は文字変数内の文字列の数を数えます。

例)

3

※ LEN関数では配列変数は使えません。

●MID\$

MID\$関数は専用文字変数(\$)に記憶されている文字列の中から、いくつかを取り出す関数です。取り出し方は、"どこから"、"いくつ"取り出すかを指定します。

例)

```

10 $="CASIO PB-400"
20 PRINT $
30 PRINT MID$(1,5)
40 PRINT MID$(7,6)
50 END

```

操 作

RUN

表 示

CASIO PB-400
CASIO
PB-400

●VAL

VAL関数は文字変数内に記憶されている数字を数値に変換します。

例)

```
10 A$="123":B$="456"
20 PRINT A$+B$
30 PRINT VAL(A$)+VAL(B$)
40 END
```

操 作

RUN 



表 示

123456
579

※ VAL関数では配列変数は使えません。

●STR\$

STR\$関数はVAL関数とは逆に、数値変数内に記憶されている数値を数字に変換します。

例)

```
10 A=147:B=258
20 PRINT A+B
30 PRINT STR$(A)+STR$(B)
40 END
```

操 作

RUN 



表 示

579
123456

例)

```
10 INPUT $
20 FOR A=1 TO LEN($)
30 PRINT MID$(A,1);
40 BEEP 1
50 NEXT A
60 END
```

このプログラムは、専用文字変数に入力された文字を、MID\$関数により1文字ずつ取り出します。取り出す位置の指定はFOR・NEXT文により決めますが、終値はLEN関数により求めます。

行番号30の最後の";"は、続けて表示させるため、表示しても止まらないようにします。

例)

```
10 A=1:B=0
20 PRINT "<";STR$(A);">";
30 INPUT $
40 IF $="END" THEN 100
50 B=B+VAL($)
60 A=A+1
70 GOTO 20
100 PRINT B/(A-1)
110 END
```


このプログラムは、データ数のわからないデータの平均を求めます。データ入力の終了は"END"を入力することにより行なわれ、行番号100に分岐して平均を表示します。

行番号20は入力をしやすくするためのメッセージで、数値変数Aをそのまま表示しますと符号桁として数字の左側が1文字分あきますので、文字としてあかないようにしています。

行番号50では、データを文字として専用文字変数に入力していますので、数値に変換して合計計算をします。

なお、このプログラムでは"END"以外の文字を入力しても、データ入力終了とはなりませんので、"END"と数値以外の入力ではエラー(ERR 2)となります。

3-3-10 覚えておく便利な入出力制御関数<KEY\$, CSR>

入出力制御関数とは、データの入力や出力を制御する関数で、"KEY\$"と"CSR"があります。

KEY\$関数は、データを入力するときに使いますが、INPUT文とは次の点異なります。

INPUT 文	KEY\$関数
●数値は仮数12桁、指数2桁以内。	●押されたキーを1文字分だけ指定された文字変数に読み込みます。
●文字は文字変数7文字、専用文字変数30文字以内。	
●"?"を表示して入力待ちとなる。	●何も表示せずに入力待ちとならない。

例)

```

10 INPUT A
20 PRINT A
30 B$=KEY$
40 IF B$="" THEN 30
50 PRINT B$
60 END

```




行番号10はINPUT文の使い方ですが、行番号30と40がKEY\$関数の使い方です。KEY\$関数はキーボードから1文字(1キー)分の入力だけを受けつけますが、キーを押しても押さなくても入力待ちで停止しません。そのため、行番号40のようにIF文と組み合わせて、キー入力があれば再び行番号30に戻り、キー入力可能な状態にします。




KEY\$関数はこのようにIF文と組み合わせて使われます。

例)

```
10 A$=KEY$
20 IF A$="1" THEN 100
30 IF A$="2" THEN 200
40 IF A$="3" THEN 300
50 GOTO 10
100 PRINT "LINE 100":END
200 PRINT "LINE 200":END
300 PRINT "LINE 300":END
```

KEY\$関数の使い方として、このような例もあります。前例ではキーが押されたかをチェックしていましたが、この例では押されたキーが1か2か3かをチェックして、合っていれば次の作業に進みます。

ただし、この例のようにKEY\$関数がプログラムの先頭の方に書き込まれているときは、プログラムのスタートの仕方に注意してください。プログラムのスタート方法として2つの方法がありますが、 のようなスタートの方法では、最後のキー操作としてキーを押しています。プログラムがスタートするとすぐにKEY\$関数でキー入力を読み込みますので、数字キーを押したままでは、そのキーが入力されてしまいます。

ためしに、プログラムエリアP1にこのプログラムを記憶させ、 として見てください。キーを少しでも長く押していると"LINE 100"が表示されます。もし、先頭でKEY\$関数を使う場合は、次のリストを加えてください。

```
5 A$=KEY$
6 IF A$="" THEN 5 } キーがはなされるまで待つ
10 A$=KEY$
  ⋮
  ⋮
```


CSR関数は、データを表示するときの位置を指定するもので、PRINT文中で使います。

例)


```
10 PRINT "A"
20 PRINT CSR 2;"A"
30 PRINT CSR 8;"A"
40 END
```

この例を実行させますと、CSR関数の働きがわかると思います。

表示上の指定は左端から0、1、2、……11となっていますので、最初のCSR関数なしでは左端から表示されますが、CSR関数で指定しますと指定した位置から表示されます。



PRINT "A"



PRINT CSR 2;"A"



PRINT CSR 8;"A"

例)

```
10 A=INT(RUN#*12)
20 PRINT CSR A;"↑"
30 GOTO 10
```

このプログラムは乱数により0～11の数値を作り出し、CSR関数で"↑"を表示させます。実行後、**EXE**キーを押すたびに"↑"が色々な所に表示されます。KEY\$関数やCSR関数を色々と組み合わせることにより、おもしろいゲームなどができます。

例)

```
10 D=0:$=" 0123456789"
20 FOR B=1 TO 10
30 IF KEY$=" " THEN 30
40 A=INT(RAN#*10)
50 PRINT MID$(1,A+1);"↑";MID$(A+3);
60 FOR C=1 TO 30
70 E$=KEY$
80 IF E$=" " THEN 100
90 NEXT C
100 IF E$<"0" THEN 130
110 IF E$>"9" THEN 130
120 IF VAL(E$)=A THEN D=D+1:BEEP 1:GOTO 140
130 BEEP 0
140 PRINT
150 NEXT B
160 PRINT CSR 2;"RIGHT";D;
170 IF D=10 THEN 210
180 FOR B=1 TO 10
190 BEEP 1:BEEP 0
200 NEXT B
210 END
```

このプログラムは0から9の数字が並んでいる中で、1文字だけ数字ではなく“↑”が出現します。この出現している場所の数字キー(②~⑨)を押すゲームです。行番号40では乱数により作り出した0～9の数値に対応する場所に“↑”を表示させています。

行番号30でのKEY\$の使い方は、キーを1度はなしてから表示させるため、前のキーを押し続けると表示は出ません。KEY\$関数はその時点のキー入力を読み込みますので、このような使い方もできます。

行番号60以降はキー入力の判断で、キーが押されれば繰り返しから抜け出し、正解かどうかをチェックします。行番号100と110は入力のチェックで、IF文により文字も比較できますので、ここでは0～9以外は行番号130に分岐させてまちがいとします。この文字の比較は137ページのキャラクター表により決っていますので、くわしくは表を参照してください。

行番号120は正解かどうかのチェックで、KEY\$では文字として読み込まれますので、VAL関数により数値に変換して比較します。

行番号30のようにキーを押しているかいないかの判断だけでは、文字変数に代入せずに判断してもかまいませんが、行番号70から120までのように何回も判断させて正解を取り出す場合は、一度文字変数に入れて使います。では、このプログラムを記憶させて、遊んでみてください。

操作例

表 示	操 作
01↑3456789	②キーを押す
01234567↑9	⑧キーを押す
0123↑56789	④キーを押す
⋮	⋮

表示のかわる速度が速すぎる方は、行番号60のFOR文の終値を30より大きくすれば、ゆっくりとキーを押せます。また、プログラムの先頭にレベルを入力するINPUT文を追加して、速度を変えてみるのもおもしろいでしょう。

3-4 あると便利なオプション

PB-400は本体だけでも、かなり便利に使えますが、オプションとしてカセットテープレコーダー用インタフェイス(アダプター)〈FA-3〉とミニプリンタ〈FP-12〉があります。

〈FA-3〉はPB-400で組んだプログラムを短時間のうちに記録したり、呼び戻したりできますし、メモリー内のデータも記録しておくことができます。

〈FP-12〉はプログラムやデータを印字するミニプリンタで、プログラムのリスト(内容)を印字して保存したり、計算結果を印字することができます。

では、各々の機能を簡単に説明していきます。

3-4-1 プログラムやデータを保存する

プログラムやデータをカセットテープに保存するには、〈FA-3〉が必要です。

PB-400とFA-3、FA-3とテープレコーダーの接続の仕方および注意事項については、FA-3に付属の取扱説明書をご覧ください。ここでは主に、使い方について説明します。

■プログラムの記録および呼び戻し

PB-400にプログラムを記憶させていくと、次のプログラムを記憶させたいが、もう入らなくなることがあります。こんなときに、前のプログラムを消してしまつては、後でもう一度使いたいときに不便です。ここでFA-3が真価を発揮します。

プログラムを記録する命令は“SAVE”または“SAVE ALL”で、“SAVE”はP0だけとか、P9だけというように、1つのプログラムエリア内のプログラムだけを記録します。“SAVE ALL”はP0からP9までの10個のプログラムエリア内のプログラムを同時に記録します。

SAVE 命令



READY P_n

└この番号のプログラムエリア内のプログラム
が記録されます。

SAVE ALL 命令

プログラムエリアに関係なく、P0からP9までの10個のプログラムエリア内のプログラムが記録されます。

SAVE 命令と SAVE ALL 命令は、単独のエリアだけのプログラムであるか、メインプログラムの他にもサブルーチンなどとしてプログラムを使っているかによって使い分けます。

SAVE、SAVE ALL 命令ともマニュアルで実行します。

例) SAVE **EXE**

SAVE "CASIO" **EXE**

SAVE ALL **EXE**

SAVE ALL "PB" **EXE**

SAVEとSAVE ALLの後に続く" "でかこんだ文字は、ファイル名と呼ばれるキーワードで、記録するプログラムに個別の名前をつけておけば、後で呼び戻すときに名前を指定して呼び戻せます。ファイル名は8文字以内のアルファベットや数字などを" "でかこんで書きます。

プログラムの呼び戻しには"LOAD"または"LOAD ALL"を使います。LOAD 命令とLOAD ALL 命令は、記録したときにSAVEまたはSAVE ALLで記録したかにより使い分けます。

記 録 \ 呼び戻し	LOAD	LOAD "ファイル名"	LOAD ALL	LOAD ALL "ファイル名"
SAVE	○	×	×	×
SAVE "ファイル名"	○	○	×	×
SAVE ALL	×	×	○	×
SAVE ALL "ファイル名"	×	×	○	○

※○印は呼び戻し可。

※ファイル名は同一のものとする。

例) LOAD **EXE**

LOAD "ファイル名" **EXE**

LOAD ALL **EXE**

LOAD ALL "ファイル名" **EXE**

LOAD 命令とLOAD ALL 命令により、プログラムを呼び戻すとき、プログラムの記録の仕方により次の表示が出ます。

記 録 方 式	表 示
SAVE	PF :
SAVE "ファイル名"	PF : ファイル名
SAVE ALL	AF :
SAVE ALL "ファイル名"	AF : ファイル名

SAVE 命令により記録したプログラムをLOAD 命令により呼び戻すとき、記録したプログラムエリアと呼び戻すプログラムエリアは同一でなくてもかまいません。

例) P0のプログラムを記録



P9に呼び戻す

〈ご注意〉

プログラムを記録したり、呼び戻すときにうまくいかないことがあります。そのようなときには次の点をチェックしてください。

- 記録しているときに“ERR 9”が表示される。

〔チェックポイント〕

PB-400とFA-3が正しく接続されているかチェックする。

- 呼び戻しているときに“ERR 9”が表示される。

〔チェックポイント〕

テープの保存状態が悪く、伸びているようなときは、新しいテープと交換する。
テープレコーダーのヘッドがよごれているときは、市販のクリーニングキットでヘッドをクリーニングする。

テープレコーダーのトーン調整を中位にする。

- 呼び戻しているときに、エラーも何も表示されずに、戻ってこない。

〔チェックポイント〕

テープレコーダーの出力ボリュームが低いので、最高(MAX)に近い位置までボリュームを上げる。

テープレコーダーの出力規格がFA-3に合わない。

■データの記録、呼び戻し

プログラムにはデータがつきものですが、いちいちキーボードから入力してはめんどうなものです。

ここでは、メモリー内のデータをテープに記録して、再び呼び戻す方法を行なってみましょう。

データを記録するには“PUT”を使います。

PUT 命令はファイル名をつけて、変数名で指定します。

PUT “ファイル名” \$, A, Z

8文字以内の文字

A～Zまでの26個の変数を記録

ファイル名はプログラムの記録と同じで、8文字以内の文字を“ ”でかこみます。変数名の指定は、専用文字変数(\$)があるときは最初に、次の2つの変数名で、“どこから”、“どこまで”の変数を記録するか指定します。

例)

専用文字変数(\$)とAからMまでの13個の変数の内容を記録する。

PUT \$,A,M

AからZ(10)までの変数の内容を、ファイル名"CASIO"で記録する。

PUT "CASIO" A,Z(10)

※メモリーが増設してある場合。

変数名の指定はどこから、どこまでのように指定しますので、"A,Z"という指定となり、"Z,A"のような指定はできません。

なお、変数を文字変数として使っている場合でも、"A\$, Z\$"とせずに"A, Z"とすることもできます。

データの呼び戻しには"GET"を使います。

GET命令もファイル名をつけて、変数名で指定します。

GET "ファイル名" \$,M,W

8文字以内の文字

M-Wまでの変数に呼び戻す

例)

専用文字変数(\$)とXからZまでの3個の変数にデータを呼び戻す。

GET \$,X,Z

G(0)からG(99)までの変数に、ファイル名"PB"のデータを呼び戻す。

GET "PB" G(0),G(99)

※メモリーが増設してある場合。

GET命令により、データを呼び戻しているとき、次のような表示が出ます。

記録方式	表示
GET \$,A,Z	DF:
GET "ファイル名" G,P	DF:ファイル名

3-4-2 記録を残す

プログラムを作るときに、プログラム内容を印字できれば、デバッグや内容変更にとっても便利です。演算結果を印字して残すのも便利なものです。

ここではミニプリンタ<FP-12>を使って印字させてみましょう。

印字させるにはプリントモード("PRT"点灯)でキー操作を行ないます。

プリントモードはMODEと押すことにより指定され、MODEと押すことにより解除されます。

プログラムの内容を印字させるには、MODEと押した後、LIST命令を実行します。

例)

次のプログラムを記録させます。

```
10 INPUT A
20 PRINT A*A
30 GOTO 10
```

記憶させた後に次の操作を行ないます。

MODE **7** **LIST** **EXE**

印 字 例

```
LIST
10 INPUT A
20 PRINT A*A
30 GOTO 10
```

もし、P0からP0までの10個のプログラムエリア全部の内容を印字させたいときは、

LIST ALL **EXE**

と操作します。

なお、印字後は**MODE** **8**と操作してプリントモードを解除してください。

演算結果等を印字させるには、**MODE** **7**と押すか、プログラム中に“MODE 7”を書き込みます。**MODE** **7**と押しますと、その後のキー操作全てが印字されますので、必要な所だけを印字させたいときは、プログラム中に書き込んだ方が便利です。

例)

演算結果だけを印字する。

```
10 INPUT A
20 MODE 7
30 PRINT A*A
40 MODE 8
50 GOTO 10
```

このプログラムでは、データ入力後にプリントモードを指定し、印字(表示もします)後プリントモードを解除して、再び入力に戻ります。

なお、プリントモード中ではPRINT文による表示は停止せずに、印字後次の命令に進みます。

MODE 7により印字後は、MODE 8でプリントモードを解除してください。

3-5 PB-100のプログラムを使う

PB-100やPB-300で作られたプログラムは、ビジネスをはじめゲームまでたくさんあり、ライブラリーも出版されています。

PB-400もPBシリーズの1つとして、この豊富なプログラムを利用できますが、PB-400にはPB-100/300以上のコマンドを持っていますので、もっと便利な使い方もできます。ここでは、PB-100/300により作られたプログラムを、PB-400で使ってみましょう。

■PB-400の相異点

PB-400とPB-100/300の使っているBASIC言語は、ほとんど共通していますが、PB-400の方が多くの命令を持っていますし、一部異なる点もあります。

●追加命令

PASS (プログラム保護)

BEEP (ブザー音)

READ (DATA文よりデータを読み込む)

DATA (データを書き込む)

RESTORE (読み込むデータを指定)

ON~GOTO (GOTO文の間接指定)

ON~GOSUB (GOSUB文の間接指定)

REM (注釈文)

●追加関数

DEG (60進数→10進数変換)

DMS\$ (10進数→60進数変換)

STR\$ (数値を数字に変換)

●変更命令

PB-400	PB-100/300
NEW(NEW ALL)	CLEAR(CLEAR ALL)
CLEAR	VAC
IF~THEN	IF~;
SAVE ALL	SAVE A
LOAD ALL	LOAD A
VERIFY	VER
DEFM(プログラム書き込み可)	DEFM(マニュアルのみ可)

●変更関数

PB-400	PB-100/300
KEY\$	KEY
MID\$	MID

以上のような相異点がありますが、PB-100/300で作られたプログラムは、原則としてそのまま使えます。

実は、PB-400にはPB-100/300で作られたプログラムを判別する機能を持っています。ただし、使いやすくするためや、後からの見直しを便利にするために、PB-400用に手直した方が良いと思います。

例)

PB-100用プログラム

```

10 VAC
20 FOR A=1 TO 20
30 INPUT Z(A)
40 IF Z(A)>80;B=B+1;GOTO 90
50 IF Z(A)<60;C=C+1;GOTO 90
60 IF Z(A)>40;D=D+1;GOTO 90
70 IF Z(A)>20;E=E+1;GOTO 90
80 F=F+1
90 NEXT A
:
:
```

この例はデータを入力して、データの大きさに応じて振り分ける部分です。このようなプログラムでも、そのまま使えますが、PB-400用のプログラムにするには、次の点を直してください。

行番号10の“VAC”は“CLEAR”にする

```
10 CLEAR
```

行番号40から70の“;”は“THEN”にする。

```
40 IF Z(A)>80 THEN B=B+1;GOTO 90
(以下同じ)
```

このプログラムでは変数の増設が必要ですので、PB-100/300ではマニュアルで実行していたDEFM命令を、プログラムの先頭に書き込みます。

```
5 DEFM 20
```

例2)

PB-100用プログラム

```
10 INPUT "I=1/O=2/P=3",N
20 IF N<1 THEN 10
30 IF N>3 THEN 10
40 GOTO N*100
  ⋮
```

このプログラムは作業に合わせて分岐先を振り分けるプログラムですが、これをPB-400に合わせるには、ON～GOTO文を用いて、次のように変更します。

```
10 INPUT "I=1/O=2/P=3",N
20 ON N GOTO 100,200,300
30 GOTO 10
  ⋮
```

このようにON～GOTO文を使うことによりスッキリできますし、判断によりデータNをチェックする必要がなくなります。

以上の内容に注意すれば、今迄作られているPB-100/300用のプログラムがより有効に使えます。

すぐにでも色々なプログラムを使ってみたい方は、市販されているPB-100/300用のプログラム集を利用できますので、とても便利です。

〔注意〕

PB-100/300で作成されたプログラムをPB-400で実行した場合に、そのままでは正しく実行しないことがあります。

(1)IF～THEN分岐先で、分岐先に数式が用いられている場合→エラー

□IF～THEN GOTO分岐先に訂正

(2)IF条件式THEN～で、条件式中で文字列の連結演算が行なわれている場合→THEN以降を実行しない。


□条件式中で文字列の連結を行なわないようにする。

1900
 1901
 1902
 1903
 1904

1905
 1906
 1907
 1908
 1909

1910
 1911
 1912
 1913
 1914
 1915
 1916
 1917
 1918
 1919
 1920

1921
 1922
 1923
 1924
 1925
 1926
 1927
 1928
 1929
 1930

A large, stylized graphic consisting of several overlapping triangles in shades of gray and black, creating a sense of depth and geometric structure. The triangles are arranged in a way that they appear to be layered, with some pointing upwards and others downwards.

第4章 コマンド・リファレンス

※ここからは、文法上の繰り返し等を説明するために、以下の記法を用います。

- 太字の語——必ず、その通り書かなければいけない語。
- $\left\{ \begin{array}{c} \times \times \times \times \\ \bigcirc \bigcirc \bigcirc \bigcirc \end{array} \right\} - \left\{ \right\}$ の中の一つを選択して書かなければなりません。
- $\left[\begin{array}{c} \bigcirc \bigcirc \bigcirc \bigcirc \end{array} \right] - \left[\right]$ の中は省略する書き方もあります。
- $\bigcirc \bigcirc \bigcirc \bigcirc *$ ——右肩に*がついた要素は繰り返して書くことができます。
- 数式——10、2+3、A、 $S * Q$ 等の数値、計算式、数値変数。
- 文字式——“ABC”、X\$, $N\$ + M\$$ 等の文字定数、文字変数、文字計算式。
- パラメータ——コマンドに伴う要素。
- (P)——プログラム中でのみ実行可能。
- (M)——マニュアルでのみ実行可能。
- (A)——プログラム中でもマニュアルでも実行可能。
- (F)——関数命令。プログラム中でもマニュアルでも実行可能。

〈例〉DATA [データ] [, [データ]]*

全ての要素に[]がついていますから、“DATA”とだけ書くこともできます。また、[, [データ]]に[]*がついていますから、この要素は繰り返して書いてもよいことになります。従って“DATA データ, データ, ……”と書くことができます。最初の[データ]の部分を省略すれば、“DATA , データ, データ, ……”と書くこともできます。

GOTO $\left\{ \begin{array}{l} \text{行番号} \\ \text{\#プログラムエリア番号} \end{array} \right\}$

これは以下のような2通りの書き方を表わしています。

- ①GOTO 行番号
- ②GOTO #プログラムエリア番号

NEW [ALL]

Ⓜ

機能

プログラムの消去。プログラムと変数の消去。

パラメータ

ALL指定したときはP0～P9の全プログラムと変数を消去します。

説明

(1)ALL指定がないときは、現在指定されているプログラムエリアのプログラムを消去します。変数は消去されません。

(2)ALL指定があるときは全プログラムエリアのプログラムと変数を消去します。

(3)パスワード設定中は実行できません。

(4)プログラム中に書き込んで使うことはできません。

(5)WRTモードでのみ実行できます。

※NEW ALLはNEW Aと省略することもできます。

例

MODE 1 NEW EXE

RUN [実行開始行]

Ⓜ

機能

プログラムの実行。

パラメータ

実行開始行：行番号

説明

(1)指定した実行開始行(省略した場合はプログラムの先頭)からプログラムを実行します。

(2)指定した行番号が存在しないときは、指定より大きく、かつ一番近い行から実行を開始します。

(3)変数はクリアされません。

例

```
10 PRINT "LINE 10"  
20 PRINT "LINE 20"  
30 END
```

RUN EXE
RUN 20 EXE

LINE 10
LINE 20

LIST $\left\{ \begin{array}{l} \text{行番号} \\ \text{ALL} \end{array} \right\}$

Ⓜ

機能

プログラムの内容を表示します。

パラメータ

行番号：表示する最初の行番号。

ALL：P0からP9までの全プログラム内容を順に表示します。

説明

I. RUNモードの場合

- (1) 行番号が指定されたときは指定行番号から、行番号が省略されたときは先頭の行から順にプログラム内容を表示します。
- (2) プログラム内容は順に自動的に表示されますので、止めたいときは[STOP]キーを押します。再び次の行以降を表示させたいときは[EXE]キーを押します。
- (3) プリンタ接続時のプリントモード("PRT"点灯中)では表示は止まらず順次、早い速度で表示します。

II. WRTモードの場合

- (1) 行番号が指定されたときは指定行番号から、行番号が省略されたときは先頭の行からプログラム内容を表示します。
 - (2) WRTモードでは1行ごとに表示して、編集可能状態となりますので、編集が必要ないときはそのまま[EXE]キーを押しますと次の行へ進みます。なお[SHIFT]キーに続けて押しますと、直前の行に戻ります。
- ALL指定があるときはP0から順にP9までの全プログラム内容を表示して行きます。このときはWRTモードでも順に送られ、編集することはできません。

- パスワード設定中は実行できません。

※LIST ALLはLIST Aと省略することもできます。

例

```
LIST [EXE]  
LIST 30 [EXE]
```


PASS “パスワード”

文字列



機能

パスワードを設定または解除します。

パラメータ

パスワード 1～8文字の文字列

説明

- (1)パスワードが設定されていないときにこの命令を実行しますと、全プログラムエリア(P0～P9)にパスワードが設定されます。
- (2)パスワードが設定されているときにこの命令を実行しますと、現在設定されているパスワードと後から実行したパスワードが一致したときのみ、パスワードが解除されます。パスワードが一致しないときはプロテクトエラー(ERR8)となります。
- (3)パスワードは1～8文字の文字列で構成され、スペース、アルファベット、数字、特殊記号などが使えます。但し、「"」自身は使えません。
- (4)パスワードが設定されているとき、LIST、LIST ALL、NEW、NEW ALLなどのコマンドは使えず、またWRTモードでの行番号`EXE`もエラー(ERR8)となり実行できません。
- (5)プログラム中では使えません。
- (6)電源スイッチオフでもパスワードは保持されます。
- (7)パスワードが設定されているときに、SAVEまたはSAVE ALL文によりプログラムをテープに記録しますと、パスワードも同時に記録されます。パスワードが設定されているプログラムを、テープからLOADまたはLOAD ALL文により読み込んだ場合は、プログラムについているパスワードが設定されます。なお、現在パスワードが設定されているときに、異なるパスワードのついたプログラムをテープから読み込むことはできません。(ERR8)

注意

パスワード設定後にパスワードを忘れてしまったときは、本体裏面のオールリセットボタンを押して、全プログラムとメモリーをクリアーしてください。

例

PASS “CASIO”`EXE`

SAVE [ALL] ["ファイル名"]

文字列

Ⓜ

機能

プログラムをカセットテープに記録します。

パラメータ

ALL: 全プログラムエリアのプログラムを記録。

ファイル名: 1～8文字の文字列。省略可。

説明

(1) ALLが省略された場合は、現在指定されているプログラムエリアの内容を記録します。

(2) ALLがついた場合は、P0からP9までの全プログラムエリアの内容を記録します。

(3) パスワードが設定されている場合は、パスワードをつけて記録しますので、LOADコマンドにより読み込んだときに同じパスワードが付きます。

※SAVE ALLはSAVE Aと省略できます。

例

SAVE **EXE**

SAVE "CASIO" **EXE**

SAVE ALL "PB" **EXE**

LOAD [ALL] ["ファイル名"]

Ⓜ

文字列

機能

プログラムをカセットテープから読み込みます。

パラメータ

ALL: 全プログラムエリアのプログラムを読み込む。

ファイル名: 1～8文字の文字列。省略可。

説明

(1)ALLが省略された場合は、現在指定されているプログラムエリアに"SAVE"で記録されたプログラムを読み込みます。

(2)ALLがついた場合は、P0からP9までのプログラムエリアに"SAVE ALL"で記録されたプログラムを読み込みます。

(3)パスワードつきで記録されたプログラムを読み込みますと、記録したときと同じパスワードが設定されます。

※LOAD ALLはLOAD Aと省略できます。

SAVEとLOADの関係

	LOAD	LOAD "ファイル名"	LOAD ALL	LOAD ALL "ファイル名"
SAVE	○	×	×	×
SAVE "ファイル名"	○	○	×	×
SAVE ALL	×	×	○	×
SAVE ALL "ファイル名"	×	×	○	○

但し、ファイル名は同一のものです。

○…読み込める

×…読み込めない

VERIFY

〔“ファイル名”〕
文字列

機能

カセットテープ上のプログラムやデータの記録状態をチェックします。

パラメータ

ファイル名：1～8文字の文字列。省略可。

説明

- (1) ファイル名が指定された場合は、同一ファイル名のファイルをチェックします。
- (2) ファイル名が省略された場合は、コマンド実行後に最初に表われたファイルをチェックします。
- (3) チェック方式は、記録状態の型式をチェックします。(パリティチェックといいます)

例

```
VERIFY EXE  
VERIFY "PROG1" EXE
```

CLEAR

Ⓐ

機能

全ての変数をクリアーします。

説明

- (1) 全ての変数をクリアーし、数値変数には0を、文字変数にはヌル(何もない)を入れます。
- (2) このコマンドはプログラム中に書き込んでも、マニュアルでも使えます。
- (3) FOR・NEXTループ(103ページ参照)中ではループ制御変数もクリアーするために、NEXT文実行時にエラーとなります。
※CLEARコマンドはVACとしても同じに使えます。

END

⑤

機能

プログラムの実行を終了します。

説明

プログラムの実行を終了しますので、次にプログラムがあっても進行しません。

STOP

⑤

機能

プログラムの実行を一時中断します。

説明

- (1)プログラムの実行を一時的に中断し、“STOP”を表示して入力待ちとなります。
- (2)中断はEXEキーにより実行を再開します。
- (3)STOP文により停止しているときに、STOPキーを押しますと、プログラムエリアと行番号を表示します。
- (4)STOPにより停止中にEXEキーによる計算が行なえます。

〔LET〕 { 数値変数=数 式 文字変数=文字式 }

⑤

機能

左辺の変数に右边の式の値を代入します。

説明

- (1)数値型変数には数値式が、文字型変数には文字式が対応します。
- (2)LETは省略することができます。

例

```
10 LET X=12
20 LET Y=X↑2+2*X-1
30 PRINT Y
40 A$="CASIO"
50 B$="PB-400"
60 PRINT A$;B$
70 END
```

REM 注 釈

Ⓟ

機 能

注釈を表わす文です。

説 明

- (1) プログラム中に書き込み、REM以降は注釈文として扱われ何も実行されません。
- (2) 同一行内に実行させたいコマンドを書き込む場合は、REM文の前にマルチステートメント(:コロン)を書いてください。

例

```
10 INPUT "R",R
20 S=π*R↑2:REM MENSEKI
30 PRINT S
40 END
```

INPUT $\left[\frac{\text{"メッセージ文"}}{\text{文字列}}, \right] \text{変数名} \left[, \left(\frac{\text{"メッセージ文"}}{\text{文字列}}, \right) \text{変数名} \right]^* \text{Ⓟ}$

機 能

キーボードからデータを変数に入力します。

パラメータ

メッセージ: 文字列。

変数名 : 数値変数名または文字変数名。

説 明

- (1) キーボードから指定した変数に入力します。
- (2) メッセージがある場合、メッセージを表示し、続けて"? "を表示します。
- (3) メッセージが省略された場合、"? "のみが表示されます。
- (4) データ入力の最後には **EXE** キーを押します。
- (5) 数値変数に文字データを入力しますとエラー(ERR2)となり、**AC** を押した後に再度"? "を表示してデータ入力を促します。
但し、アルファベット1文字や数式を入力した場合、数式の結果が数値のときはその値が代入されます。

例

```
10 INPUT A
20 INPUT "B$=",B$
30 INPUT "C$=",C$,"D$=",D$
```

KEY\$

Ⓟ

機能

キーボードから1文字を入力する関数です。

説明

(1) キーボードからのキー入力を1文字分だけ受け入れます。

(2) 数字、アルファベット、記号がキー入力できます。

(3) 読み込まれたデータは1文字の文字型となります。

(4) “?”は表示せず、入力待ちにもなりませんので、通常はIF文と組み合わせて使います。

※KEY\$はKEYと省略することもできます。

例

```
10 PRINT "INPUT<6>";  
20 A$=""  
30 K$=KEY$  
40 IF K$="" THEN 30  
50 A$=A$+K$  
60 IF LEN(A$)<6 THEN 30  
70 PRINT A$  
80 END
```

● 6文字をキーボードから受けつけます。

PRINT [出力要素] [{ ' } [出力要素]] * ^(P)

機能

出力要素を表示します。

パラメータ

出力要素：出力制御関数(CSR)、数式、文字式。

説明

(1)出力要素を表示します。出力制御関数がつく場合は、それによって決められた位置から表示します。

(2)数式、文字式ではその値を表示します。

(3)出力要素が数式の場合は、値の前に符号桁(+, -)がつきますが、+符号は空白として表示されます。

●文字型表示——
 出力要素


●数値型表示——
 符号 出力要素

(4)出力要素が数式で仮数部が10桁以上の場合は、11桁目を四捨五入して表示します。また、仮数部以外に符号桁と指数部があるときは指数記号(E)と指数部2桁を表示します。

(5)出力要素がメッセージ等の文字定数の場合は、" "の間の文字数は30文字以内です。

(6)出力要素と出力要素の区切りは,と;が使える,のときは前の出力要素を表示後停止し(STOP点灯)、**EXE**キーにより、一度表示をクリアーしてから、次の出力要素を表示します。区切りが;のときは前の出力要素に続けて次の出力要素を表示します。

(7)出力要素が全て省略されたとき(PRINTのみ)は、表示をクリアーするだけで停止はしません。

(8)**MODE**  と押すプリントモードで印字中は、PRINT文を実行しても表示は停止しません。

(9)SET文により数値をフォーマット化することができます。

例

```
10 PRINT 1/3
20 PRINT "A=" ; A
30 PRINT "SIN 30" , SIN 30
40 PRINT "END" ;
50 PRINT
60 END
```


CSR 出力位置指定 数 式

Ⓕ


機 能 指定した位置から出力要素を表示します。

パラメータ 出力位置指定：数式で、値は小数以下切り捨てとなります。

$$0 \leq \text{指定} < 12$$

説 明 (1) PRINT文中で用い、出力要素の出力位置を指定します。

(2) 出力位置の与え方は左端を 0 とします。



例

```
10 FOR I=0 TO 11
20 PRINT CSRI;"A";CSR 11-B;"B"
30 NEXT I
40 END
```

● A と B の文字が **EXE** キーを押すごとに左右から移動します。

GOTO

{ 分岐先行番号
行番号
プログラムエリア番号
0～9の1文字 }

Ⓟ

機能

指定された分岐先へ無条件で分岐します。

パラメータ

分岐先行番号 : 1～9999の行番号
プログラムエリア番号 : 0～9の1文字

説明

(1) 指定された分岐先へ分岐します。

(2) 分岐先が行番号の場合は、現在のプログラムエリア内の指定行へ分岐し、プログラムを実行します。分岐先行番号が存在しない場合は、エラー(ERR 4)となります。

(3) 分岐先がプログラムエリア番号の場合は、指定されたプログラムエリアへ分岐し、先頭からプログラムを実行します。

※ 分岐先行番号、プログラムエリア番号に数式も使えます。

例

```
10 PRINT "START";  
20 GOTO 100  
30 PRINT "LINE 30"  
40 END  
100 PRINT "LINE 100"  
110 END
```

ON

分岐条件 数 式

GOTO

[分岐先][, [分岐先]]* (P)

分岐先は { 分岐先行番号
 # プログラムエリア番号

機 能

分岐条件に従って指定された分岐先へ分岐します。

パラメータ

分岐条件 : 数式で、値は小数以下切り捨てとなります。

分岐先行番号 : 1~9999の行番号

プログラムエリア番号: 0~9の1文字

説 明

(1) 分岐条件の式の値の整数部により分岐します。分岐先は先頭から順に式の値が1の場合、2の場合……と割り当てられます。

ON A GOTO $\frac{100}{A=1}, \frac{200}{A=2}, \frac{300}{A=3}, \dots\dots$

(2) 式の値が1より小さいか、または相当する分岐先が書いてないときは分岐せずに、次の文を実行します。

(3) 分岐先は一行に納まるまで、いくつでも書けます。

例

```
10 INPUT A
20 ON A GOTO 100,200,300
30 PRINT "OTHER"
40 GOTO 10
100 PRINT "LINE 100":GOTO 10
200 PRINT "LINE 200":GOTO 10
300 PRINT "LINE 300":GOTO 10
```

● 1~3を入力すると100~300行に分岐し、それ以外では
"OTHER"を表示します。

IF 分岐条件 THEN

比較式

{文[:文]}*
{分岐先}

★分岐先は {分岐先行番号
#プログラムエリア番号}

Ⓟ

機能

分岐条件が成立したとき、THEN以降の文を実行します。また THEN以降が分岐先の場合は分岐します。

パラメータ

分岐条件 : 比較式
分岐先行番号 : 1~9999の行番号
プログラムエリア番号 : 0~9の1文字

説明

(1) 分岐条件が成立したとき、THEN以降の文を実行または分岐先へ分岐します。

(2) 分岐条件が成立しなかったときは、次の行を実行します。

(3) 分岐条件は比較式(=、≠、<、>、≤、≥)により判断します。

= 左辺と右辺が等しい ≠ 左辺と右辺が等しくない
< 左辺より右辺が大きい > 左辺より右辺が小さい
≤ 左辺より右辺が大きい ≥ 左辺より右辺が小さい
か等しい か等しい

(4) 比較式には、文字列の連結は使えません。

(5) 2つ以上分岐条件がある場合は、THENの後にIF文を続けることができます。

IF~THEN IF~THEN.....

※ THENの後が文の場合は、THENのかわりに;が使えます。

例

```
10 N=6
20 PRINT CSR N;"↑";
30 K$=KEY$
40 IF K$="4" THEN N=N-1:IF N<0 THEN N=0
50 IF K$="6" THEN N=N+1:IF N>11 THEN N=11
60 PRINT
70 GOTO 20
```

● "↑"が[4]キーを押すと左に、[6]キーを押すと右に動きます。

FOR 制御変数名 = 初期値 数式 TO 終値 数式 (STEP 刻み幅 数式) (P) NEXT 制御変数名

機能

FOR文からNEXT文までの間を制御変数を初期値から終値まで刻み幅で変化させながら繰り返します。

パラメータ

制御変数名：単純変数名で、配列変数は使えません。

初期値：数式

終値：数式

刻み幅：数式。省略したときは1の値

説明

(1) FOR文からNEXT文までの間を制御変数を初期値から終値まで、刻み幅で変化させながら繰り返し、制御変数が終値を超えたとき繰り返しを終了します。

(2) 初期値が終値を超えている場合は、FOR～NEXTの間を1度だけ実行します。

(3) 刻み幅は負数も使え、省略した場合は1となります。

(4) FOR文とNEXT文は必ず1対1で対応していなければなりません。また、FOR文に対応するNEXT文は、FOR文より後に書きます。

(5) FOR～NEXTのループは次のように入れ子構造にすることができます。

```
10 FOR I=1 TO 10
20 FOR J=11 TO 20
30 PRINT I, J
40 NEXT J
50 NEXT I
60 END
```

(6) 入れ子構造にすることをネスティングともいい、4重までできます。

(7) FOR～NEXTループを終了したとき、制御変数は終値を超えたときの値となります。

(8) FOR～NEXTループからの外への飛び出しは可能ですが、IF文、GOTO文などでループ内へ飛び込むとエラーになります。なお、ループから飛び出したときも、ループの中であることを記憶していますので、NEXT文で終了させない限りネスティングを重ねていきます。

GOSUB

{ 分岐先行番号
行番号
プログラムエリア番号
0 ~ 9 の 1 文字 }

Ⓟ

機 能

指定された分岐先へサブルーチン分岐します。

パラメータ

分岐先行番号 : 1 ~ 9999 の行番号

プログラムエリア番号 : 0 ~ 9 の 1 文字

説 明

- (1) 指定された分岐先へサブルーチン分岐します。サブルーチンからの戻りはRETURNの実行により行なわれます。
- (2) サブルーチンの中からさらにサブルーチンを引用することをネスティングといい、8段までできます。
- (3) RETURNによりGOSUB文の次の文に戻ります。
- (4) GOSUB文によりサブルーチン分岐したときに、IF文やGOTO文などで次の文に戻すと、ネスティングは記憶されたままです。必ずRETURNで戻してください。
- (5) 分岐先行番号が存在しない場合はエラー(ERR 4)となります。
※分岐先行番号、プログラムエリア番号に数式も使えます。

例

```
10 PRINT "MAIN 10"  
20 GOSUB 100  
30 PRINT "MAIN 30"  
40 END  
100 PRINT "SUB 100"  
110 GOSUB 200  
120 RETURN  
200 PRINT "SUB 200"  
210 RETURN
```

RETURN

Ⓟ

機 能

サブルーチンから復帰します。

説 明

サブルーチンを呼んだ直後の文へ復帰します。

ON 分岐条件 数式 GOSUB [分岐先]([, [分岐先]))*

Ⓟ

★分岐先は $\begin{cases} \text{分岐先行番号} \\ \text{井プログラムエリア番号} \end{cases}$

機能

分岐条件に従って指定された分岐先のサブルーチンへ分岐します。

パラメータ

分岐条件 : 数式で、値は小数以下切り捨てとなります。

分岐先行番号 : 1~9999の行番号

プログラムエリア番号 : 0~9の1文字

説明

(1) 分岐条件の式の値の整数部によりサブルーチン分岐します。

分岐先は先頭から順に式の値が1の場合、2の場合……と割り当てられます。

ON B GOSUB $\frac{1000}{B=1}$, $\frac{2000}{B=2}$, $\frac{3000}{B=3}$ ……

(2) 式の値が1より小さいか、または相当する分岐先が書いてないときは分岐せずに、次の文を実行します。

(3) 分岐先は一系列に納まるまで、いくつでも書けます。

例

```
10 INPUT A
20 ON A GOSUB 100,200,300
30 GOTO 10
100 PRINT "SUB 100":RETURN
200 PRINT "SUB 200":RETURN
300 PRINT "SUB 300":RETURN
```

●1~3を入力すると各々のサブルーチンへ分岐します。

DATA

[データ](,データ)*

定数

定数

Ⓟ

機能

データを収納します。

パラメータ

データ：文字定数または数値定数

説明

- (1) READ文で読み取るデータを書く為に用います。
- (2) データは、で区切って複数個書くことができます。
- (3) データを省略すると長さ0の文字列を表わします。
- (4) データ文だけを実行しても何もしません。
- (5) 文字定数の中に、を含むときは、データの両端を"で囲んでください。

DATA ABC, DEF, "GHI,JKL",

1つ目

2つ目

3つ目

READ 変数名[, [変数名]]*

Ⓟ

機能

DATAの内容を読み取ります。

パラメータ

変数名：数値変数または文字変数。配列変数も可。

説明

- (1)現在指定されているDATA文の中のデータを順に割り当て、指定された変数に代入します。
- (2)数値変数には数値型のデータのみ読み取れます。
- (3)DATA文の中のデータは、行番号の小さい方から大きい方へ、また同一DATA文中では先頭から順に読まれます。
- (4)READ文で必要な数を読んだ後、次のREAD文で読まれるのは、そのさらに後にあるデータです。
- (5)プログラムの動作の初めには、どのデータも指定されません。READ文の最初の実行で、そのREAD文のあるプログラムエリアの先頭のデータが読まれ、以後はこのときのプログラムエリアのデータが順に読まれていきます。
- (6)RESTORE文により読み込むデータの指定を変えることができます。
- (7)READ文の変数よりDATA文中のデータが少ないときはエラー(ERR4)となります。
- (8)DATA文中のデータの先頭にスペースがあるときは読みとばします。

例

```
10 DATA 1,2,3
20 READ A,B
30 PRINT A;B
40 DATA 4,5
50 READ C,D,E
60 PRINT C;D;E
70 END
```

- DATA文から順にデータを読み込み、表示します。

RESTORE

〔行番号〕

数式

Ⓟ

機能

READ文で読むデータの位置を指定します。

パラメータ

行番号：数式で、値は小数以下を切り捨てとなります。

$$1 \leq \text{行番号} \leq 9999$$

説明

- (1) READ文で読むデータのあるDATA文を指定します。
- (2) 行番号を省略すると、データの指定を解除します。この後最初に実行するREAD文により、そのREAD文のあるプログラムエリアの先頭にあるデータが指定され、読めます。
- (3) 行番号を指定すると、RESTORE 文が存在するプログラムエリアの行番号が指定されます。以後のREAD文では、そのときのプログラムエリアのデータが次々に読めます。
- (4) 指定された行番号が存在しないときや指定された行番号以降にDATA文が存在しないときは、エラー(ERR4)となります。

例

```
10 DATA 1,2,3
20 DATA 4,5
30 READ A,B,C,D,E
40 RESTORE 10
50 READ F,G
60 RESTORE 20
70 READ H,I
80 PRINT A;B;C;D;E;F;G;H;I
90 END
```

PUT ["ファイル名"]変数[, 変数]*

文字列

Ⓐ

機能

カセットテープにデータを記録します。

パラメータ

ファイル名：1～8文字の文字列。省略可。

変数 ：記録する変数範囲の指定

説明

(1)カセットテープに変数の内容を記録します。

(2)変数の指定は次のように書きます。

PUT A変数Aの内容

PUT A,Z変数A～Zの内容

PUT A,A(100)変数A～A(100)の内容

PUT \$,D,W専用文字変数\$とD～Wの内容

専用文字変数\$の内容を記録する場合は\$を最初に書き込みます。

(3)マニュアルでもプログラム中に書き込んでも使えます。

GET ["ファイル名"]変数[, 変数]*

文字列

Ⓐ

機能

カセットテープに記録されているデータを変数に読み込みます。

パラメータ

ファイル名：1～8文字の文字列。省略可。

変数 ：読み込む変数の指定

説明

(1)カセットテープに記録されているデータを指定された変数に読み込みます。

(2)変数の指定は次のように書きます。

GET A変数Aに読み込む

GET A,Z変数A～Zに読み込む

GET A,A(100)変数A～A(100)に読み込む

GET \$,D,W専用文字変数\$とD～Wに読み込む

(3)PUTにより記録した変数名とGETにより読み込む変数名は一致していなくてもかまいません。

(4)読み込もうとする変数より記録されているデータが少ない場合は、記録されているデータだけ、先頭の変数から順に読み込みます。

(5)マニュアルでもプログラム中に書き込んでも使えます。

BEEP $\left\{ \begin{smallmatrix} 0 \\ 1 \end{smallmatrix} \right\}$

Ⓐ

機能

ビーブ音を発生させます。

パラメータ

0 : 低音。

1 : 高音。

省略した場合は0と同じ。

説明

(1) 低い音と高い音のビーブ音を発生させます。

(2) マニュアルでも使えます。

例

```
10 $="ABCDEFGHIJKLMNOPQRSTUVWXYZ":N:0
20 FOR I=1 TO 10
30 A$=MID$( RAN#*26+1,1 )
40 PRINT CSR4;"<";A$;">";
50 FOR J=1 TO 30
60 K$=KEY$: IF K$#"" THEN 80
70 NEXT J
80 IF K$=A$ THEN BEEP 1:N=N+1:GOTO 100
90 BEEP 0
100 PRINT:NEXT I
110 PRINT:N;
120 IF N>10 THEN END
130 FOR I=1 TO 10
140 BEEP 0:BEEP 1
150 NEXT I
```

●表示された文字に対応するアルファベットキーを押します。

DEFM [増設メモリー数]

Ⓐ

数 式

機 能

メモリーの増設をします。

パラメータ

増設メモリー数：数式で、値は小数以下切り捨てとなります。
省略可。 $0 \leq \text{増設メモリー数} < 453$ 。

説 明

- (1)メモリー(変数エリア)の増設を行ないます。
- (2)増設メモリー数は1個単位で残りステップ数に応じて任意に指定できます。
- (3)メモリー1つ増設するごとに8ステップ必要となりますので、プログラムエリアのステップ数は減少します。
- (4)配列変数を使用し、データメモリーを多く必要とするときに使用します。
- (5)増設メモリー数を省略した場合は、現在設定されているメモリー数を表示します。
- (6)マニュアルでもプログラム中に書き込んでも使え、マニュアルで実行した場合は新しい設定状態(増設メモリー数+基本メモリー数26)を表示します。プログラム中に書き込んで実行した場合は新しい設定状態は表示されません。
- (7)残りステップ数よりも多くのメモリーを増設しようとした場合は、エラー(ERR1)となります。

例

DEFM 10 **EXE**

***VAR:36

DEFM **EXE**

***VAR:36

—WRTモードでは残りステップ数を表示します。

```
10 DEFM 10
20 FOR I=1 TO 10
30 INPUT Z(I)
40 NEXT I
```

⋮

MODE 数式

★数式は 4 ~ 8

Ⓟ

機能

計算機の状態を設定します。

パラメータ

数式：値は小数以下切り捨てとなります。

$$4 \leq \text{数式} < 9$$

説明

(1) 数値の値により角度単位やプリントモードの設定・解除を設定します。

(2) 設定は次の通りです。

MODE4.....角度単位を度に設定します。

MODE5.....角度単位をラジアンに設定します。

MODE6.....角度単位をグラジアンに設定します。

MODE7....."PRT"を表示し、プリントモードに設定します。

MODE8.....プリントモードを解除します。

(3) これは **MODE** キーによる設定と同じですが、RUNモードやWRTモードの設定はできません。

例

```
10 MODE 4
20 A=SIN 30
30 MODE 7
40 PRINT A
50 MODE 8
60 END
```

SET

$$\begin{Bmatrix} Fn \\ En \\ N \end{Bmatrix}$$

★ n は0～9の整数

Ⓐ

機能

数値データの出力形式(フォーマット)を指定します。

パラメータ

Fn : 小数点以下指定。

En : 有効桁数指定。

N : 指定解除。

説明

- (1) 数値データを出力するときの、小数点以下の桁数や有効桁数の指定を行いません。
- (2) 小数点以下指定(Fn)では、小数点以下の桁数を0から9桁まで指定します。
- (3) 有効桁数指定(En)では、有効桁数を1から10桁まで指定します。なお、“SET E0”としたときは10桁指定となります。
- (4) “SET N”では両指定を解除します。
- (5) マニュアルでもプログラム中に書き込んでも使えます。

例

```
10 INPUT N
20 SET F5:PRINT N
30 SET E5:PRINT N
40 SET N:GOTO 10
```

LEN (単純文字変数)

⑥

機能

与えられた単純文字変数の中の文字数を値とする関数です。

パラメータ

単純文字変数：配列変数は使えません。

説明

(1)単純文字変数の中の文字数を数える関数です。

(2)使える文字変数は単純文字変数(A\$, Y\$等)で、配列文字変数(B\$(3)等)は使えません。

例

```
10 INPUT A$  
20 PRINT LEN(A$)  
30 GOTO 10
```


MID\$ (位置[, 文字数])

Ⓕ

機能

専用文字変数(\$)の指定した位置から指定文字数を取り出す関数です。

パラメータ

位置 : 数式で、値は小数点以下を切り捨てます。

$1 \leq \text{位置} < 101$

文字数 : 数式で、値は小数点以下を切り捨てます。

$1 \leq \text{文字数} < 101$

省略すると、位置以降の全てが指定されます。

説明

(1) 専用文字変数(\$)の指定した位置から指定した文字数分の文字を取り出す関数です。

(2) 位置が文字列の範囲を超えて指定されたときは、ヌル (何もない)を与えます。

(3) 文字列の、指定した位置以降の長さが指定した文字数より小さいときは、指定した位置以降の文字列全部を取り出します。

※MID \$はMIDと省略することができます。

例

```
10 $="ABCDEFGHIJKLMN OPQRSTUVWXYZ"  
20 INPUT M,N  
30 PRINT MID$(M,N)  
40 END
```

VAL (単純文字変数)

⑥

機能

単純文字変数内の数字を数値に変換する関数です。

パラメータ

単純文字変数：配列変数は使えません。

説明

(1) 単純文字変数内の数字を数値に変換します。

(2) 文字変数の内容が数字のみ(＋、－、・、E、E⁻を含む)の場合は、そのまま数値に変換します。

A\$="12.3"のとき、VAL(A\$)→12.3

(3) 文字変数の内容が数字以外で始まっている場合は、エラーとなります。

A\$="A45"のとき、VAL(A\$)→エラー(ERR 2)

(4) 文字変数の内容が数字で始まっているが、途中に数字以外が入っている場合は、最初の数字部分を数値に変換します。

A\$="78A9"のとき、VAL(A\$)→78

例

```
10 INPUT A$
20 PRINT VAL(A$)
30 END
```

STR\$ (数式)

⑥

機能

数式の値を文字(数字)に変換します。

パラメータ

数式：数値、計算式、数値変数、配列数値変数

説明

(1) 数式の値を文字に変換します。

(2) 数式が計算式の場合は、計算結果を文字に変換します。

例

```
10 PRINT STR$(123)
20 PRINT STR$(45+78)
30 A=963
40 PRINT STR$(A)
50 END
```

数値関数

SIN 引数
数式

COS 引数
数式

TAN 引数
数式

Ⓕ

機能

与えられた引数に対する三角関数の値を与える関数です。

パラメータ

引数：数式。

$-1440^\circ < \text{引数} < 1440^\circ$ (度)

$-8\pi < \text{引数} < 8\pi$ (ラジアン)

$-1600 < \text{引数} < 1600$ (グラジアン)

但し、**TAN**においては $|\text{引数}| = (2n-1) * 1$ 直角を除く

$1 \text{ 直角} = 90^\circ = \frac{\pi}{2} \text{ rad} = 100 \text{ grad}$

説明

(1)与えられた引数に対する三角関数の値を与える関数です。

(2)値は角度単位の設定 (**MODE** キー、モードコマンド)に従います。

ASN 引数
数式

ACS 引数
数式

ATN 引数
数式

Ⓕ

機能

与えられた引数に対する逆三角関数の値を与える関数です。

パラメータ

引数：数式。ASN、ACSは $-1 \leq \text{引数} \leq 1$ 。

説明

(1)与えられた引数に対する角度を与える逆三角関数です。

(2)値は角度単位の設定 (**MODE** キー、モードコマンド)に従います。

(3)関数の値は以下の範囲で与えられます。

$-90^\circ \leq \text{ASN } X \leq 90^\circ$

$0^\circ \leq \text{ACS } X \leq 180^\circ$

$-90^\circ \leq \text{ATN } X \leq 90^\circ$

LOG 引数 数式

LN 引数 数式

Ⓕ

機能

対数関数の値を与える関数です。

パラメータ

引数：数式。 $0 < \text{引数}$ 。

説明

対数関数の値を与えます。

● LOG 常用対数関数 $\log_{10} x$ 、 $\log x$

● LN 自然対数関数 $\log_e x$ 、 $\ln x$

EXP 引数 数式

Ⓕ

機能

指数関数の値を与える関数です。

パラメータ

引数：数式。 $-227 \leq \text{引数} \leq 230$ 。

説明

指数関数の値を与える関数です。

EXP e^x

SQR 引数 数式

Ⓕ

機能

引数の平方根を与える関数です。

パラメータ

引数：数式。 $0 \leq \text{引数}$ 。

説明

引数の平方根を与える関数です。

SQR \sqrt{x}

ABS 引数

Ⓕ

数式

機能

引数の絶対値を与える関数です。

パラメータ

引数：数式。

説明

引数の絶対値を与える関数です。

ABS $|x|$

SGN 引数

Ⓕ

数式

機能

引数の符号に応じた値を与える関数です。

パラメータ

引数：数式。

説明

引数の符号に応じた値を与える関数です。

引数が正の場合、1

引数が0の場合、0

引数が負の場合、-1

INT 引数

Ⓕ

数式

機能

引数を越えない最大の整数を与える関数です。

パラメータ

引数：数式。

説明

引数を越えない最大の整数を与える関数です。

INT 12.56→12

INT -78.1→-79

FRAC 引数

数式

Ⓕ

機能

引数の小数部を値とする関数です。

パラメータ

引数：数式。

引数の小数部を値とする関数です。符号は引数の符号と一致します。

RND (引数, 桁位置)

数式

数式

Ⓕ

機能

引数を指定した桁で四捨五入した値を与える関数です。

パラメータ

引数：数式。

桁位置：数式。値は小数点以下切り捨てとなります。

$-100 < \text{桁指定} < 100$

説明

(1) 引数を指定した桁で四捨五入した値を与える関数です。

(2) 桁指定は10の桁位置乗の位置を四捨五入します。

小数点以下3桁目を四捨五入 $\rightarrow \text{RND}(x, -3)$

100の位を四捨五入 $\rightarrow \text{RND}(x, 2)$

RAN#

Ⓕ

機能

0 から 1 の間の乱数を与える関数です。

説明

(1) 0 から 1 の間の乱数を与える関数です。 $0 < \text{乱数} < 1$

(2) 乱数は小数点以下10桁です。

例

0 ~ 9 の1桁の乱数を作る

$\text{INT}(\text{RAN}\# * 10)$

1 ~ 5 の1桁の乱数を作る

$\text{INT}(\text{RAN}\# * 5) + 1$

10 ~ 99 の2桁の乱数を作る

$\text{INT}(\text{RAN}\# * 90) + 10$

DEG (度[,分[,秒]])

Ⓕ

機能

60進数を10進数に変換します。

パラメータ

度：数式。

分：数式。

秒：数式。

$|\text{DEG}(\text{度}, \text{分}, \text{秒})| < 10^{100}$

説明

度・分・秒で示される60進数を10進数に変換します。

例

$\text{DEG}(12, 34, 56)$ **EXE**

12.58222222

10 INPUT A,B,C

20 PRINT DEG(A,B,C)

30 END

DMS\$ (引数)

数式

Ⓕ

機能

10進数を60進数に変換します。

パラメータ

引数：数式。|数式|<10¹⁰⁰

説明

(1)10進数を60進数に変換します。

(2)変換された結果は、文字列として与えられます。

例

DMS\$(45.678) **EXE**

45°40'40.8

10 INPUT A

20 \$=DMS\$(A)

30 PRINT\$

40 END

第5章 ライブラリー

二次回帰分析

このプログラムは、データ x と y を二次回帰分析して、 y の推定値を求めます。
計算は $y = ax^2 + bx + c$ に基づいて、

$$a = \frac{\sum(x^2 \cdot y) \cdot \sum(x \cdot x) - \sum(x \cdot y) \cdot \sum(x \cdot x^2)}{\sum(x \cdot x) \sum(x^2 \cdot x^2) - \{\sum(x \cdot x^2)\}^2}$$

$$b = \frac{\sum(x \cdot y) \sum(x^2 \cdot x^2) - \sum(x^2 \cdot y) \sum(x \cdot x^2)}{\sum(x \cdot x) \cdot \sum(x^2 \cdot x^2) - \{\sum(x \cdot x^2)\}^2}$$

$$c = \frac{\sum y_i}{n} - b \frac{\sum x_i}{n} - a \frac{\sum x_i^2}{n}$$

として求めます。

P0に入れるプログラムはデータ入力プログラムで、データ x とデータ y を入力し、同時に $\sum x$ 、 $\sum y$ 、 $\sum x^2$ 、 $\sum xy$ 、 $\sum x^3$ 、 $\sum x^4$ 、 $\sum x^2 \cdot y$ を求めます。

P1に入れるプログラムでは、P0のプログラムにより求められた $\sum x$ 、 $\sum y$ 、 $\sum x^2$ 、 $\sum xy$ 、 $\sum x^3$ 、 $\sum x^4$ 、 $\sum x^2 \cdot y$ から a 、 b 、 c を計算し、続いて入力された x に対する推定値 y を求めます。

では、実際に次の例を使って計算してみましょう。

例)

	1	2	3	4	5	6	7
x	1	5	8	11	15	18	22
y	21	30	41	54	70	①	②

このデータを回帰分析して、①と②の推定値を求めます。

操 作

SHIFT P0

1 EXE

21 EXE

5 EXE

以下順にデータを
入力します。

70 EXE

表 示

x1?
y1?
x2?
y2?
⋮
y5?
x6?

データ入力が終わりましたら、次にP1のプログラムを実行します。

操 作

SHIFT P1
EXE
EXE
EXE
18 EXE
EXE
22 EXE

表 示

a = 0.08967a
b = 2.14288b
c = 18.23781c
x = ?	
y = 85.86245 y の推定値
x = ?	
y = 108.78103 y の推定値

このようにP1のプログラムでa、b、cとyの推定値を求めます。

ここでは答を小数以下5桁まで求めています、丸める必要がないときや、他の桁で丸めたいときはP1の行番号90を変換してください。

メモリー	
A	Σx^3
B	Σx^4
C	$x^2 \cdot y$
I	a
J	b
K	c
N	Σx
M	データ数
O	Σy
P	Σx^2
R	$\Sigma x \cdot y$

P0

```
10 CLEAR
20 PRINT "x";H+1;:
  INPUT X
30 PRINT "y";H+1;:
  INPUT Y
40 N=N+X:O=O+Y
50 P=P+X↑2:R=R+X*Y
60 A=A+X↑3:B=B+X↑4
  :C=C+X↑2*Y
70 M=M+1
80 GOTO 20
```

P1

```
10 D=P-N↑2/M
20 E=R-N*O/M
30 F=A-N*P/M
40 G=C-P*O/M
50 H=B-P↑2/M
60 I=(G*O-E*F)/(D*
  H-F↑2)
70 J=(E*M-G*F)/(D*
  H-F↑2)
80 K=O/M-J*M/H-1*P
  /M
90 SET F5
100 PRINT "a=";I,"b
  =";J,"c=";K
110 INPUT "x=";X
120 Y=I*X↑2+J*X+K
130 PRINT "y=";Y
140 GOTO 110
```

計309ステップ

万能たてよこ集計

このプログラムは6個の独立したプログラムからできています。

P0に入れるプログラムは「データ入力プログラム」で、表のたてとよこを指定して、データを入力します。

P1に入れるプログラムは「表示(印字)プログラム」で、表の中のデータを順次表示または印字します。

P2に入れるプログラムは「データ編集用プログラム」で、一度記憶させたデータの訂正を行ないます。

P3に入れるプログラムは「計算プログラム」で、たて計とよこ計、そして総合計を求めます。

P4に入れるプログラムは「テープ記録用プログラム」で、メモリー内のデータをカセットテープに記録します。

P5に入れるプログラムは「テープからの読み込み用プログラム」で、カセットテープに記録してあるデータをメモリー内に読み込みます。

では、次のデータを使って操作してみましょう。

	1	2	3	4	5	6
1	376	159	248	767	311	351
2	320	85	287	833	291	541
3	480	41	166	750	426	367
4	518	269	343	565	221	268
5	536	158	426	495	235	492

操 作

SHIFT **P0**

Y

5 **EXE**

6 **EXE**

376 **EXE**

159 **EXE**

248 **EXE**

...

235 **EXE**

492 **EXE**

表 示

New[Y/N]?
TATE?
YOKO?
(1, 1)?
(1, 2)?
(1, 3)?
(1, 4)?
...
(5, 6)?
END

...新たに表を作るときはYを押します。

...たての項目数を入力します。

...よこの項目数を入力します。

...順番にデータを入力します。

...データ入力の終了です。

次に、入力したデータを確認してみましょう。

操 作	表 示	
SHIFT P1	Printer[Y/N]?	…プリンタに印字するときはYを押します。
N	(1, 1) 376	…以下、EXEを押すごとにデータ内容を表示していきます。
EXE	(1, 2) 159	
EXE	(1, 3) 248	
⋮	⋮	
EXE	(5, 6) 492	
EXE	END	

P2に入れた編集用のプログラムは、入力したデータがまちがっていたときや、一部のデータを変更して計算をしたいときに使います。

では、たてが3で、よこが4のデータを“450”とまちがえていたとします。

操 作	表 示	
SHIFT P2	TATE?	…たての位置を指定します。
3 EXE	YOKO?	…よこの位置を指定します。
4 EXE	(3, 4) 450?	…たて3、よこ4のデータを表示します。
750 EXE	(3, 5) 462?	…すので訂正が必要なときは、正しい数値を入力します。

以後、次のデータを見たいときは **+ EXE** と押し、前のデータを見たいときは **= EXE** と押します。これで、前後のデータも訂正できます。

+ EXE	(3, 6) 367?
+ EXE	(4, 1) 518?

訂正が終了したら **= EXE** と押せば、最初の“TATE?”に戻り、たてとよこの指定になりますし、“TATE?”と表示されているときに **= EXE** と押せば、このプログラムは終了します。

= EXE	TATE?
= EXE	END

なお、データが表示されているときに、数値を入力しますと新たな数値の入力となりますので、注意してください。

P3に入れたプログラムは、たて計とよこ計、そして総合計を求めます。

操 作	表 示	
[SHIFT] P3	Printer[Y/N]?	…プリンタに印字するときはYを押します。
N	YOKO TOTAL	…まず、よこ計を表示します。
[EXE]	(1) 2212	
[EXE]	(2) 2357	
⋮	⋮	
[EXE]	TATE TOTAL	…次に、たて計を表示します。
[EXE]	(1) 2230	
[EXE]	(2) 712	
⋮	⋮	
[EXE]	GRAND TOTAL	…最後に総合計を表示します。
[EXE]	011325	

P4とP5に入れたプログラムはデータ保存用で、カセットインタフェース<FA-3>をお持ちの方で、入力したデータを保存したいときに使用します。

P4のプログラムは記録用で、プログラムスタートと同時にデータを記録しますので、スタート前に<FA-3>とカセットテープレコーダに接続し、マイク用端子とリモート用端子を差し込んで、「録音」状態にしておいてください。

P5のプログラムは再生用ですので、スタート前に、<FA-3>とカセットテープレコーダーに接続し、イヤホン用端子とリモート端子を差し込み、「再生」状態にしておいてください。

なお、記録するときは新しいテープを、再生するときはデータの記録してあるテープをセットしておいてください。

<ポイント>

このプログラムでは全部で1,091ステップをプログラムとして使用していますので、データ数(たて×よこ)は315個以内となっています。もし、もっと多くのデータを扱いたい場合は、不要なプログラムを記憶させずに、残りステップ数に応じてP0の行番号80の“315”を変更してください。

P3に入れている計算用のプログラムは、たて計とよこ計、そして総合計を求めるものですので、他の計算を行ないたい方は、このプログラムを変更してみるのもよいでしょう。

P0

```

10 PRINT "New [Y/N
J?";
20 K$= KEY$: IF K$
="Y" THEN PRINT
: GOTO 50
30 IF K$="" THEN 2
0
40 PRINT : GOTO 21
0
50 CLEAR
60 INPUT "TATE",Y
70 INPUT "YOKO",X
80 IF Y*X>315 THEN
60
90 DEFN X*Y
100 FOR I=1 TO Y
110 FOR J=1 TO X
120 PRINT "(";I;",";
J;")";: INPUT
$
130 IF $>"*" THEN I
F $<"x" THEN 18
0
140 IF $*="" THEN 1
10
150 IF J-1>0 THEN J
0J-1: GOTO 120
160 IF I-1<1 THEN 1
20
170 I=I-1:J=X: GOTO
120
180 Z((I-1)*X+J)= Y
AL($ )
190 NEXT J
200 NEXT I
210 PRINT "END"

```

271ステップ

P1

```

10 PRINT "Printer[
Y/N]";
20 K$= KEY$: IF K$
="" THEN 20
30 PRINT
40 IF K$="Y" THEN
MODE 7: PRINT "
DATA"
50 FOR I=1 TO Y
60 FOR J=1 TO X
70 PRINT "(";I;",";
J;")";Z((I-1)*
X+J)
80 NEXT J
90 IF K$="Y" THEN
PRINT " "
100 NEXT I
110 MODE 8
120 PRINT "END"

```

151ステップ

P2

```

10 INPUT "TATE", $
20 IF $="" THEN P
RINT "END";: EN
D
30 IF $>"*" THEN I
F $<"x" THEN 50
40 GOTO 10
50 INPUT "YOKO",P
60 O= VAL($ )
70 PRINT "(";O;",";
P;")";Z((O-1)*
X+P);: INPUT $
80 IF $="" THEN 1
0
90 IF $="+" THEN 1
40
100 IF $="-" THEN 1
60
110 IF $>"*" THEN I
F $<"x" THEN 13
0
120 GOTO 70
130 Z((O-1)*X+P)= Y
AL($ )
140 IF P+1>X THEN O
=O+1:P=0: IF O>
Y THEN O=1:P=1:
GOTO 70
150 P=P+1: GOTO 70
160 IF P-1<1 THEN O
=O-1:P=X+1: IF
O<1 THEN O=Y:P=
X: GOTO 70
170 P=P-1: GOTO 70

```

293ステップ

P3

```

10 PRINT "Printer[
Y/N]";
20 K$= KEY$: IF K$
  =" " THEN 20
30 IF K$="Y" THEN
  MODE 7
40 PRINT
50 PRINT "YOKO TOT
  AL"
60 FOR I=1 TO Y
70 A=0
80 FOR J=1 TO X
90 A=A+Z((I-1)*X+J
  )
100 NEXT J
110 PRINT "(*;I;)"
  ;A
120 NEXT I
130 PRINT "TATE TOT
  AL"
140 B=0
150 FOR J=1 TO Y
160 A=0
170 FOR I=1 TO X
180 A=A+Z((I-1)*X+J
  )
190 NEXT I
200 PRINT "(*;J;)"
  ;A
210 B=B+A
220 NEXT J
230 PRINT "GRAND TO
  TAL"
240 PRINT B
250 MODE 8

```

263ステップ

P4

```

10 PRINT "DATA PUT
  ";
20 PUT "DATA"X,Y
30 PUT Z(1),Z(X*Y)
40 PRINT "→END"

```

53ステップ

P5

```

10 PRINT "DATA GET
  ";
20 GET "DATA"X,Y
30 DEFN X*Y
40 GET Z(1),Z(X*Y)
50 PRINT "→END"

```

60ステップ

計 1091ステップ

算数博士

このプログラムは、たし算、ひき算、かけ算、わり算の問題を出します。

四則の選択の他に、1桁か2桁の計算が選べます。

問題は10問で、1問につき10点が与えられます。

なお、わり算の答は小数点以下2桁まで(切り捨て)で採点します。

操 作

RUN **EXE**

表 示

Push + - * /

…たし算は $+$ を、ひき算は $-$ を、かけ算は $*$ を、わり算は $/$ を押す。

+

1

12 **EXE**

9 **EXE**

13 **EXE**

1ch or 2ch

… 1桁か2桁を選択

(1) 3+9=?

… 1問目

(2) 2+7=?

… 2問目

(3) 2+6=?

… 3問目

(10) 6+7=?

… 10問目

SCORE: 100

… 点数表示

答が正しいときは“ピッピッ”と鳴ります。

100点満点のときは点数表示後、ファンファーレのように鳴ります。

```
10 E=0
20 PRINT "Push + -
  * /";
30 K$= KEY$
40 IF K$="+" THEN
  M=1: GOTO 90
50 IF K$="-" THEN
  M=2: GOTO 90
60 IF K$="*" THEN
  M=3: GOTO 90
70 IF K$="/" THEN
  M=4: GOTO 90
80 GOTO 30
90 BEEP 1
100 PRINT : PRINT "
  1ch or 2ch";
110 K$= KEY$
120 IF K$="1" THEN
  N=9: GOTO 150
130 IF K$="2" THEN
  N=99: GOTO 150
140 GOTO 110
150 BEEP 1
```

```
160 FOR I=1 TO 10
170 PRINT
180 A= INT( RAN#*N+
  1)
190 B= INT( RAN#*N+
  1)
200 ON M GOSUB 300,
  400,500,600
210 PRINT "<"; STR$
  (I);"> ";$;: IN
  PUT D
220 BEEP 0:D= INT(D
  *100)/100
230 IF C=D THEN E=E
  +1: BEEP 1: BEE
  P 1
240 NEXT I
250 PRINT "SCORE:";
  E*10;
260 IF E=10 THEN FO
  R I=1 TO 10: BE
  EP 1: BEEP 0: N
  EXT I
```

```
270 END
300 $= STR$(A)+"+"+
  STR$(B)+"="
310 C=A+B
320 RETURN
400 IF A<B THEN F=A
  :A=B:B=F
410 $= STR$(A)+"-"+
  STR$(B)+"="
420 C=A-B
430 RETURN
500 $= STR$(A)+"x"+
  STR$(B)+"="
510 C=A*B
520 RETURN
600 IF A<B THEN F=A
  :A=B:B=F
610 $= STR$(A)+"÷"+
  STR$(B)+"="
620 C= INT(A/B*100)
  /100
630 RETURN
```

ベルトコンベアーゲーム

このゲームはベルトコンベアーによって運ばれてくる製品をよりわけます。
製品は良品"○"と不良品"×"とがあり、自分の位置"■"に製品がきたときに不良品を取り出します。

不良品は自分の位置に"×"がきたとき、**+**キーを押します。

良品はそのまま運ばせます。

不良品を取り出すと2ポイント、良品を運ぶと1ポイント与えられます。

なお、3回失敗するとゲームオーバーとなります。

操作例

RUN/EXE

表示

```
BELT CONVEYOR
>X-----■<
>_X-----■<
```

…まず不良品がきました。

(今だ！)

+

```
>-----X_<
>-----*<
```

…正しく不良品を取り出した場合は
"×"が"*"にかわります。

+

```
>-----■X<
<<DAME!!>>
```

…押すのが遅れた。

…失敗です。

+

```
<<DAME!!>>
[GAME-OVER]
SCORE 7
```

…3回目の失敗です。

…ゲームオーバーです。

…7点でした。

```

10 DEFN 0
20 $="BELTCONVEYOR
   ": GOSUB 500
30 D=0:M=0
40 $="xox"
50 C= INT( RAN#*3)
   +1
60 G= INT( RAN#*10
   )
70 Z= INT( RAN#*10
   +1)
80 FOR A=1 TO 11:
   PRINT CSR0;">__
   -----<" CSR
   Z;"||";
90 K$= MID(C,1)
100 IF C=2 THEN IF
   A=11 THEN D=D+1
   : BEEP 1: BEEP
   1: GOTO 40
110 IF A=11 THEN 26
   0
120 PRINT CSR0;K$;:
   BEEP 1: BEEP 0
130 I$= KEY$: IF I$
   ="+" THEN 180
140 FOR B=1 TO 6: N
   EXT B
150 PRINT CSR0;" ";
160 REM IF A=11 THE
   N 40
170 NEXT A
180 IF C=2 THEN 260
190 IF A*Z THEN 220
200 D=D+2: PRINT CS
   RA;"*";
210 FOR I=1 TO 5: B
   EEP 0: BEEP 1:
   NEXT I
220 IF A*Z THEN 260
230 FOR F=0 TO 70
240 NEXT F
250 GOTO 40
260 PRINT CSR0;" ((
   DAME! ))";
270 FOR E=0 TO 5
280 BEEP 0: BEEP 0:
   BEEP 1
290 NEXT E
300 M=M+1: IF M<3 T
   HEN 40
310 $="[GAME-OVER]
   ": GOSUB 500
320 PRINT CSR0;"SC0
   RE";D$: BEEP 0:
   END
500 PRINT
510 FOR H=1 TO LEN(
   $)
520 PRINT MID$(H,1)
   : BEEP 1
530 NEXT H
540 FOR H=0 TO 40:
   NEXT H
550 PRINT
560 RETURN

```


計 513ステップ

並びかえゲーム

このゲームは不規則に並んでいる0から9の数字を正しく"0123456789"と並びかえます。

使うキーは[4]と[6]と[+]で、[4]を押すとカゴが左に移動し、[6]を押すとカゴが右に移動します。[+]キーを押すと、現在のカゴの位置にある数字をカゴに入れ、カゴに入っていた数字をかわりに置きます。

表示の見方



*の点滅がカゴの位置です。

表示されるSCORE(点数)は作業時間で、小さい程良いのです。

なお、ハイスコアが残るようになっていきますので、最初に行うときや、他のプログラムを実行した後は、必ず"CLEAR^{EXE}"と操作してください。

操 作

CLEAR ^{EXE}
RUN ^{EXE}

[+]
[6][6][6][6]
[+]
[4][4][4]
[+]

表 示

	…最初だけ行ないます
Hi-Sc:0	…ハイスコア表示
<<WAIT!>>	…問題作成中
3721*94650>	
3721*94650>8	…8をカゴに入れる
3721 946*0>8	…正しい位置に移動
3721 946*0>5	…を置く。5がカゴに入る
3721 *4680>5	…5を置く。9がカゴに入る
3721 *4680>9	…正しい位置に移動

以下順に繰り返す

[+]

0123*56789>4	…最後の4を置く
[0123456789]	…並びかえ成功
SCORE 132	…スコア表示

ゲームを続ける場合は、直接"RUN^{EXE}"と操作してください。

カゴに入れる順番やキーを押すタイミングにより、スコアはどんどん縮まります。

★あなたの腕まえは？

スコアー	評 価
50未満	あなたは天才!! 運も強い。
50～69	抜群の反射神経です。
70～99	やっ和人並み。もう一歩です。
100～149	まだまだ練習不足です。
150以上	何かのまちがいでしょう? もう一度どうぞ。

```

10 PRINT "Hi-Sc:";
   H:= FOR I=1 TO
10: BEEP 1: NEX
   T I
20 $=""
30 PRINT : PRINT "
   << WAIT ! >>";
40 A$= STR$( INT(
   RAN#*10))
50 FOR I=1 TO LEN(
   $)
60 IF A$= MID$(I,1
   ) THEN 40
70 NEXT I
80 $=$+A$
90 IF LEN($)<10 TH
   EN 40
100 PRINT
110 X=5:B$="" :N=0
120 PRINT CSR0;$;">
   ";B$;
130 PRINT CSRX;"*";
140 K$= KEY$
150 IF K$="4" THEN
   BEEP 0:X=X-1: I
   F X<0 THEN X=0
160 IF K$="5" THEN
   BEEP 0:X=X+1: I
   F X>9 THEN X=9
170 IF K$="+" THEN
   GOSUB 500
180 N=N+1
190 IF $*"012345678
   9" THEN 120
200 IF H=0 THEN H=N
210 IF H>N THEN H=N
220 PRINT : PRINT "
   [";$;"1";
230 FOR I=1 TO 10
240 BEEP 0: BEEP 1
250 NEXT I
260 PRINT
270 PRINT "SCORE";N
   ;
280 END
500 BEEP 1
510 IF X=0 THEN C$=
   MID$(1,1):$=B$
   + MID$(2): 60TO
   540
520 C$= MID$(X+1,1)
530 $= MID$(1,X)+B$
   + MID$(X+2)
540 B$=C$
550 RETURN

```

エラーメッセージ一覧表

エラーコード	意 味	エ ラ ー 原 因	対 策
1	メモリーオーバーまたはシステムスタックオーバー	<ul style="list-style-type: none"> ●ステップ数不足でプログラムが書き込めない。またはメモリーが増設できない。 ●計算式が複雑すぎてスタックオーバーしている。 	<ul style="list-style-type: none"> ●不要なプログラムをクリアするか、メモリー数を減らす。 ●数式を分割して簡単にする。
2	構文エラー	<ul style="list-style-type: none"> ●プログラム等に書式上の誤りがある。 ●代入文等で左辺の型式と右辺の型式が異なる。 	<ul style="list-style-type: none"> ●入力したプログラム等の誤りを修正する。
3	数学的エラー	<ul style="list-style-type: none"> ●数式の演算結果が10^{100}以上の場合。 ●数値関数の入力範囲外の場合。 ●結果が不定または不能となる場合。 	<ul style="list-style-type: none"> ●演算式またはデータを修正する。 ●データを判断する。
4	未定義エラー	<ul style="list-style-type: none"> ●GOTO文、GOSUB文の指定行番号がない。 ●READ文に対するデータが不足している。 	<ul style="list-style-type: none"> ●指定行番号を修正する。 ●データ数を正しくする。
5	引数エラー	<ul style="list-style-type: none"> ●引数を必要とするコマンド、関数において、引数が入力範囲外の場合。 	<ul style="list-style-type: none"> ●引数の誤りを修正する。
6	変数エラー	<ul style="list-style-type: none"> ●増設されていないメモリーを使おうとした。 ●同一メモリーを数値変数と文字変数に同時に使おうとした。 	<ul style="list-style-type: none"> ●適切にメモリーを増設する。 ●同時に同一メモリーを文字変数、数値変数として使わないようにする。
7	ネスティングエラー	<ul style="list-style-type: none"> ●サブルーチン実行中以外でRETURN文が出てきた場合。 ●FORループ中以外でNEXT文が出てきたり、FOR文に対するNEXT文の変数が異なる場合。 ●サブルーチンのネスティングが8段をこえた場合。 ●FORループのネスティングが4段をこえた場合。 	<ul style="list-style-type: none"> ●不必要なRETURN文やNEXT文を取る。 ●サブルーチンやFOR・NEXTループをレベル内にする。

エラーコード	意味	エラー原因	対策
8	パスワードエラー	<ul style="list-style-type: none"> ●パスワードが設定されているときに、異なるパスワードを入力した。 ●パスワードが設定されているのに、LISTやNEWなどの禁止コマンドを実行した。 	<ul style="list-style-type: none"> ●正しいパスワードを入力して、パスワードを解除する。
9	オプションエラー	<ul style="list-style-type: none"> ●テープレコーダが接続されていないのに、SAVEまたはPUTコマンドを実行した場合。 ●LOADまたはGETコマンドにより入力された信号がわれており、読み込めない場合。 ●プリンタの充電が十分でない場合。 ●プリンタの紙づまり。 	<ul style="list-style-type: none"> ●テープレコーダを接続する。 ●テープレコーダの再生ボリュームを下げる。 ●テープレコーダのTONEを中位に調整する。 ●テープをかえる。 ●テープレコーダのヘッドをクリーニングする。 ●プリンタを充電する。 ●プリンタの紙づまりを直す。







キャラクター表






	スペース	+	-	*	/	↑	↓	"	#	\$	>	≥	=	≤	<	キ
数字	0	1	2	3	4	5	6	7	8	9	.	π)	(E	E
英大文字	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P
	Q	R	S	T	U	V	W	X	Y	Z						
英小文字	a	b	c	d	e	f	g	h	i	j	k	l	m	n	o	p
	q	r	s	t	u	v	w	x	y	z						
記号	?	,	;	:												
グラフィック記号	○	Σ	◦	△	@	×	÷	♠	←	♥	♦	♣	μ	Ω	↓	→
	%	¥	□	[&	_	'	•]	■						

表は左から右へ、次の段の左から右へと小さい順に並んでいます。

一番小さいキャラクターは、スペースで、一番大きいキャラクターは、グラフィック記号の■(拡張モードにし、**[SHIFT][Z]**を押すと表示される)です。

フローチャートの主な記号

記号	名称	意味
	端子	開始、終了等
	手操作入力	キーボードからのデータ入力
	出力	出力の機能
	処理	あらゆる種類の処理機能
	定義済み処理	サブルーチンなど、別の場所で定義されている命令群
	判断	いくつかの択一的経路のうち、どの経路をとらせるかを決める判断

記 号	名 称	意 味
	FOR・NEXTループ	FOR文とNEXT文の間で、一定回数分処理を行なう。
	書 類	書類を媒体とする入出力機能
	流 れ 線	記号を結びつける機能
	給 合 子	フローチャートのほかの場所への出口またはほかの入口
	注 釈	明瞭にするため、説明または注意を加える機能

コマンド索引

ABS	18, 119	REM	96
ACS	18, 117	RESTORE	67, 108
ASN	18, 117	RETURN	57, 104
ATN	18, 117	RND	18, 120
BEEP	65, 110	RUN	36, 89
CSR	74, 99	SAVE(ALL)	78, 92
CLEAR	94	SGN	18, 119
COS	18, 117	SIN	18, 117
DATA	67, 106	SQR	18, 118
DEFM	64, 111	STOP	43, 95
END	95	STR\$	72, 116
EXP	18, 118	TAN	18, 117
FOR~TO~STEP/NEXT	54, 103	VAL	72, 116
FRAC	18, 120	VERIFY	94
GET	80, 109		
GOSUB	57, 104		
GOTO	45, 100		
IF~THEN	49, 102		
INPUT	25, 96		
INT	18, 119		
KEY\$	74, 97		
LEN	72, 114		
LET	28, 95		
LIST	90		
LOAD(ALL)	78, 93		
LOG	18, 118		
MID\$	72, 115		
NEW(ALL)	89		
PASS	91		
PRINT	47, 98		
PUT	80, 109		
RAN#	18, 121		
READ	67, 107		

規 格

型 式：PB-400

基本計算機能：負数、指数、カッコを含む四則計算(加減・乗除の優先順位判別機能つき)

組込関数機能：三角・逆三角関数(角度単位は度・ラジアン・グラジアン)、対数・指数関数、開平、べき乗、整数化、整数部除去、絶対値、符号化、有効桁数指定、小数点以下指定、 $10 \leftrightarrow 60$ 進変換、乱数、 π

コ マ ン ド：INPUT、PRINT、GOTO、ON~GOTO、FOR~NEXT、IF~THEN、GOSUB、ON~GOSUB、RETURN、READ、DATA、RESTORE、STOP、END、RUN、LIST、LIST ALL、MODE、SET、CLEAR、NEW、NEW ALL、DEFM、PASS、REM、LET、SAVE、SAVE ALL、LOAD、LOAD ALL、PUT、GET、VERIFY

プログラム関数：KEY\$、CSR、LEN、MID\$、VAL、STR\$

計 算 範 囲： $\pm 1 \times 10^{-99}$ ~ $\pm 9.999999999 \times 10^{99}$ および 0、内部演算は仮数部12桁を使用

プログラム言語：BASIC(ベーシック)

R A M 容 量：約4.3Kbyte

(システムエリア約0.5Kbytes、メモリーエリア約240bytesを含む)

ス テ ッ プ 数：最大3616ステップ

組込プログラム数：最大10組(P0~P9)

メ モ リ ー 数：標準26~最大478メモリー、および専用文字変数(\$)

ス タ ッ ク 数：サブルーチン 8 段・FOR~NEXTループ 4 段
数 定 値 6 段・演 算 子 12 段

表示方式および：仮数部10桁(負符号含む)、または仮数部8桁(負数7桁)+指数部2桁、内容 EXT、 \boxed{S} 、RUN、WRT、DEG、RAD、GRA、TR、PRT、STOPの各状態表示付

表 示 素 子：12桁ドットマトリクス液晶

主 要 素 子：C-MOS VLSI他

電 源：リチウム電池(CR-2032) 2 個使用

消 費 電 力：最大0.03W

電 池 寿 命：本体のみ170時間
(連続使用)

オートパワーオフ：約7分

使 用 温 度： 0°C ~ 40°C

大きさ・重さ：幅165 奥行71 高さ9.8mm、130g(電池込み)

付 属 品：ソフトケース

カシオ計算機株式会社営業本部

東京都新宿区西新宿2-6 新宿住友ビル
(〒160) ☎03-347-4811(代表)

カシオ計算機サービスセンター

旭川	0166-23-8580	〒070	旭川市七条通り8丁目
札幌	011-231-2343	〒060	札幌市中央区南一条西12丁目
釧路	0154-24-8575	〒085	釧路市光陽町6-1-12
青森	0177-22-7466	〒030	青森市勝田2-1-15
秋田	0188-33-6211	〒010	秋田市中通り6-1-15
盛岡	0196-24-2502	〒020	盛岡市本町通り3-19-6
仙台	0222-27-1404	〒980	仙台市一番町2-3-32
山形	0236-42-8018	〒990	山形市あこや町3-14-39
山形	0249-33-5172	〒963	福島県郡山市香久池2-16-6
宇都宮	0286-34-0395	〒320	宇都宮市西大寛2-1-3
前橋	0272-53-3000	〒371	前橋市元総社町92-5
水戸	0292-25-6985	〒310	水戸市中央1-2-20
埼玉	0486-66-8567	〒330	大宮市大成町1-1-8
千葉	0472-43-1751	〒260	千葉市登戸町2-2-7
東京	03-862-4141	〒101	千代田区神田佐久間町2-23
中央	03-583-4111	〒106	港区六本木2-3-6
大田	03-787-3721	〒145	大田区上池台1-1-6
新宿	03-376-3221	〒160	新宿区西新宿4-2-18
立川	0425-23-3531	〒190	立川市錦町3-2-25
横浜	045-211-0821	〒231	横浜市中区弁天通り6-85
新潟	0252-41-4105	〒950	新潟市米山3-1-5
長野	0262-28-9360	〒380	長野市岡田町30-20
甲府	0552-37-6371	〒400	甲府市城東2-22-11
静岡	0542-81-8085	〒420	静岡市西中原1-4-35
浜松	0534-64-1658	〒435	浜松市西塚町3-2-4
豊橋	0532-53-2515	〒440	豊橋市魚町5
名古屋	052-263-0454	〒460	名古屋市中区栄4-6-15
岐阜	0582-62-0145	〒500	岐阜市鷹見町8
津市	0592-27-5066	〒514	津市鳥居町1-9-1
富山	0764-22-2251	〒930	富山市白銀町2-1-1
金沢	0762-37-8511	〒920	金沢市諸江町下丁93-1
京都	075-351-1161	〒600	京都市下京区五条通り堀川東入ル
大阪	06-362-8181	〒530	大阪市北区南森町2-1-20
和歌山	0734-31-7807	〒640	和歌山市九家の丁5
神戸	078-392-4123	〒650	神戸市中央区北長狭通り4-4-18
岡山	0862-41-8471	〒700	岡山市西古松西町8-21
福山	0849-24-2830	〒720	福山市南本庄町2-1-3
広島	082-263-1090	〒730	広島市南区稲荷町4-1
山口	0835-22-6164	〒747	防府市戎町1-10-16
高松	0878-62-5240	〒760	高松市亀岡町9-16
松山	0899-45-2234	〒790	松山市平和通り1-1-5
福岡	092-411-2684	〒812	福岡市博多区博多駅南1-2-15
長崎	0958-61-8084	〒852	長崎市宝栄町2-26
熊本	0963-67-0650	〒862	熊本市健軍4-1-5
鹿児島	0992-56-3575	〒890	鹿児島市上荒田町30-18



CASIO®